
Abstract

Specifications of sufficient conditions may be enough for reasoning about complete and

unchanging programs of a closed system. Nevertheless, there is no luxury of trusting

external components of probably unknown provenance in an open world that may be

buggy or potentially malicious. It is critical to ensure that our components are robust

when cooperating with a wide variety of external components. Holistic specifications,

which are concerned with sufficient and necessary conditions, could make programs more

robust in an open-world setting.

In this thesis, we lay the foundations for reasoning about holistic specifications. We

give an Isabelle/HOL mechanization of holistic specifications focusing on object-based

programs. We also pave a way to reason about holistic specifications via proving some key

lemmas that we hope will be useful in the future to establish a general logic for holistic

specifications.

Chapter 1

Introduction

Traditional system designs are often implicitly based on a closed world assumption where

a component can trust to interact with other component’s operations: a client who sup-

plies arguments meeting that operation’s pre-conditions can invoke it and obtain the

associated effect. Nevertheless, a system could be more complex, buggy, or potentially

malicious when it collaborates with a wide range of external components. Such a system

is considered to deal with open-world settings. Indeed, in an open world, we do wish to

trust the other components with which we collaborate. Trusting our personal information

to software that operates in an open world might make us vulnerable to hackers and

exploits. As a result, there is a need for our software to be robust. We expect that software

to perform correctly, even if used by external parties of probably unknown provenance,

buggy, or potentially malicious. For instance, medical patients expect their health data

not to be sent to their employer(s) unless they authorized the release.

There are numerous studies on the specification and verification of programs’ functional

correctness [15, 29, 19, 13, 1]. Most of these methods based on pre- and post-conditions

are rooted in design-by-contract assumptions: “If the pre-condition is not satisfied, the

routine is not bound to do anything” [19]. Such specifications describe what sufficient

conditions are for some effect to happen. This approach is enough to reason about com-

plete, unchanging programs of closed systems. However, in an open world, things are

more complicated; systems must deal with a range of external components that might

not be under our control. Since methods of external components cannot control when

they are invoked, we must work as if all externally visible methods have the pre-condition

true. Sufficient conditions are not adequate in open-world settings; therefore, there is a

need to have the necessary conditions to ensure that bad things will not happen. To do

that, Drossopoulou et al. [8] first introduced holistic specifications, which are kinds of

specifications dealing with sufficient conditions as well as necessary conditions. They,

namely, proposed a specification language, named Chainmail, to present holistic specifi-

cations. In addition to traditional specification languages, the design of Chainmail draws

concepts from object capabilities [21], temporal logic, and spatial connection. Thanks

to these features, Chainmail can express holistic specifications for several examples not

only from object-capability programs but also the smart contracts applications such as

Bank/Account example [22] and DAO (Decentralised Autonomous Organisation) [9], a fa-

mous Ethereum contract aiming to provide smart contracts, managed by the DAO owners

and not affected by a central government.

Although a formal specification is more precise than a natural language, it could disagree

or contain errors with requirements. Verified formal specifications, in this case, are needed

when we desire to be sure that the specification follows its requirements. With the same

goal, we want to have a verified formal specification framework for Chainmail. We choose

Isabelle [26], a higher-order logic (HOL) theorem prover, to give Chainmail a verified

formal specification. Therefore, our primary goal is to lay the foundations for reasoning

about holistic specifications. To have a skeleton for reasoning about holistic specifica-

tions, we give a formalization of holistic specifications focusing on object-based programs

Isabelle in this thesis. Mainly, we provide an Isabelle mechanization of Chainmail with

several definitions, theorems, and technical lemmas. Moreover, to pave a way to reason

about holistic specifications, we prove some key lemmas that we hope will help establish

a general logic for holistic specifications.

The rest of this thesis is structured as follows. We present the background of Isabelle/HOL,

holistic specifications, and survey related work in Chapter 2. Then, we describe our main

contribution to Chapter 3 and 4. We address future work and conclude the thesis in

Chapter 5. Finally, Appendix A describes several technical definitions, functions, and

lemmas supporting the formalization of holistic specifications and presents a partial proof

of a theorem mentioned in Chapter 4.

1.1 Contributions

The main contributions of this thesis are the following:

1. We have given the first version of the formalization of Chainmail in Isabelle and

provide several lemmas related to the holistic concepts.

2. We have built lemmas and provide proofs and “pen-and-paper” proofs 1 to place

the foundations for reasoning about holistic specifications.

1This thesis will distinguish between “proofs” done in Isabelle and “pen-and-paper proofs” sketched but
not verified in Isabelle.





Chapter 2

Background and Related Work

This chapter introduces the minimum necessary preliminaries of holistic specifications

and Isabelle/HOL to follow the subsequent sections. Furthermore, we present a survey on

current work related to the object-capability model [21], focusing on specification and

verification for object-capability programs.

Chapter Outline

• Background. Section 2.1 gives a brief overview of Isabelle/HOL and its details of the

necessary technical background, covers definitions and feature concepts of holistic

specifications, and presents Bank/Account example.

• Related Work. Section 2.2 provides surveys on the foundations and current publica-

tions related to the object-capability model.

2.1 Background

2.1.1 Isabelle/HOL

Isabelle/HOL [26] is a proof assistant, a computer program that assists in conducting

proofs of theorems using higher-order logic. It is being developed at the University of

Cambridge and Technische Universität München. Isabelle/HOL gives the languages for

mathematical reasoning and the rules similar to natural deduction’s rules to carry out

proofs. We can utilize Isabelle/HOL to show mathematical proofs and reason and prove

the semantics of a programming language and its properties. Isabelle/HOL developments

comprise a list of theories, the definition of functions, types, sets, and a set of lemmas,

theorems, and so forth, interpreted by the theorem prover. It also provides two ways of

writing proofs: (1) a tactic script and (2) a structured proof language called Isar (Intelligible

Semi-Automated Reasoning). In this thesis, we frequently shift between tactic script and

Isar to produce proofs of lemmas.

Furthermore, Isabelle/HOL supplies us with three means to validate whether our theories

or lemmas are correct or not. These are, respectively, counter-example commands, tactics,

and inference tools. The commands quickcheck and nitpick are used to search for and

generate counter-examples. Tactics include auto and metis and assist us in proving spe-

cific goals automatically. Other automated tools, such as sledgehammer or solve_direct,

help us create proofs for the current goal by recommending tactics.

We now provide in detail the necessary technical background on Isabelle/HOL to under-

stand this thesis’s remainder. First, we want to talk about Types. They are (1) basic types,

in particular nat, the type of natural numbers, (2) type constructors, particularly list,

the type of lists, (3) arbitrary types represented by variable (denoted by ’a ), (4) recursive

types represented by the datatype command introducing recursive data types.

Moreover, the keyword typedecl describes a new type name without any additional

assumptions, e.g., typedecl FieldName describes a set of field names FieldName. Besides,

the keyword type_synonym introduces a synonym for the type specified, e.g., type_synonym

Identifier = nat introduces a synonym Identifier for the type of natural number nat.

Datatype option is used to add a new element None to an existing type ’a. For instance, in

this thesis, we usually use the option type to represent partial functions. In particular, the

partial function from a to b is represented by the type a ⇒ b option.

Second, we speak about Terms: (1) function application f t is the call function f with

t is an argument, (2) function abstraction λx. t is the function, where x is a parameter

and returning value t.

The third is a definition, also called a non-recursive definition. It is defined as a definition

command, for instance, inc as follows.

definition inc:: "nat ⇒ nat"

where

"inc n = n + 1"

Besides, we use the recursive function in this thesis. We most use two common ways

to define a recursive function: (1) fun defines a more expressive function, and we might

need to prove termination manually, and (2) primrec, a restrictive version of fun, defines

a recursive function in which we need to state every rule. Furthermore, the inductive

definition is essential. Thus, it is the key construct of operational semantics in the next

part of the thesis.

Forth, records are used in the thesis, generalizing tuples’ concept, but their components

have names instead of position. A record declaration introduces new types and types of

abbreviations. For example, the record of type pt represents a point in three-dimensional

space, in which three fields named x, y, and z of type int.

record pt = x :: int y :: int z :: int

Finally, as we mentioned earlier, in this thesis, there are two ways of writing proofs: a tactic

script and an Isar. Let us give a glimpse at both of them. A tactic script, called "apply" style

proofs, is backward reasoning, progressing from goal to premises. On the other hand, Isar

appears like a mathematical reasoning style with structured proofs and similar notations

such as assume, have, thus, and hence.

We give a toy example that proves P ∧ Q −→ Q ∧ P, to say the difference between "apply"

and Isar.

In the first one applyStyle utilizing the apply tactic, the audience will probably face

difficulty to read since the proof does not show the result of each step.

lemma applyStyle: "P ∧ Q −→ Q ∧ P"

apply (rule impI)

apply (rule conjI)

apply (rule conjunct2)

apply assumption

apply (rule conjunct1)

apply assumption

done

On the other hand, the second one, IsarStyle, is more structured and readable. Similar

to the mathematical language, we want to show intermediate steps as statement G1 and

statement G2. Then, from these intermediate steps, we can prove the ultimate goal.

lemma IsarStyle: "P ∧ Q −→ Q ∧ P"

proof

assume H: "P ∧ Q"

from H have G1: "P"

by (rule conjunct1)

from H have G2: "Q"

by (rule conjunct2)

from G2 and G1 show "Q ∧ P"

by (rule conjI)

qed

2.1.2 Holistic Specifications

Specifications of functional correctness of programs describe what sufficient conditions

for some effect to happens are. Considering a bank system as an example, if we have

enough money and make a payment request to the bank, the money will be transferred,

and as a consequence, our balance will be reduced. Enough money and the payment

request are a sufficient condition for the decrease in money. In contrast, necessary condi-

tions guarantee that things will not happen. For instance, we desire the bank to ensure no

reduction in our bank balance will occur unless requested. The difference between suffi-

cient and necessary conditions is that sufficient conditions are described on an individual

function. Necessary conditions, on the other hand, are about the behavior of a component

as a whole. When our component cooperates with other unknown provenance systems

in an open world, we want the component to meet its sufficient and necessary conditions.

Consequently, its specifications should be holistic. Namely, it illustrates the overall be-

havior of a component: each function’s behavior and limitations on the behavior that

emerges from combinations of functions. In other words, holistic specifications [8] must,

therefore, address sufficient as well as necessary conditions. The discrepancy between

classical invariants and holistic specifications is that the classical invariants reflect on

the current program state. In contrast, holistic specifications reflect on all aspects of a

program’s execution, possibly over all the components making up that program.

Holistic specifications extend traditional program specifications with ideas from object

capabilities (permission and authority), temporal logic, spatial connectives, and viewpoint,

describing in [8] as follows.

• Permission: Objects may have access to which other objects; this is fundamental as

access to an object privileges access to its functions.

• Control: Objects called functions on other objects; this is beneficial in identifying

the causes of specific effects.

• Time: What holds some time in the past, the future, and what changes with time.

• Space: Which parts of the heap are considered when establishing some property or

when performing program execution.

• Viewpoint: What objects and runtime configurations 2 are internal to our compo-

nent, which are external to it.

Chainmail [8] is a specification language to draw the holistic specifications addressed

beforehand. Mainly, Chainmail assertions include pure expressions, classical assertions

about the contents of heap and stack, comparisons between expressions, and the usual

logical connectives. Also, Chainmail can address holistic ideas, including permission,

control, time, space, and viewpoint.

• Permission: Permission states an object has a direct path to another object, repre-

sented as assertion 〈x Access y〉 saying that if there is a direct path from the object

x to another object y: either object x and y are aliases, or object x points to an object

with a field whose value is the same as the object y, or object x is currently executing

an object, and object y is a local parameter.

• Control: Control assertion represents the object making a function call on another

object, expressed as assertion 〈x Calls y.m(zs)〉 stating that it holds if in a run-

time configuration in which there is a method on the object x that performs the

method call y.m(zs). Here, object x is a caller, object y is a receiver, and x calls

method m on y with arguments zs.

• Time: Temporal assertions are a part of the holistic assertions consisting of Next〈A〉,
Will〈A〉, Prev〈A〉, and Was〈A〉. In particular, assertions Next〈A〉 and Will〈A〉 talk

about the future, in which A holds at the immediate successor step, and some future

point, respectively. Otherwise, assertions Prev〈A〉 and Was〈A〉 say about the past, in

which A holds at the predecessor step and a number of steps in the past, respectively.

• Space: An assertion 〈A In S〉 says that A holds in a runtime configuration restricted

to objects from a given set S. In other words, the objects that make A valid should be

included in the set S.

• Viewpoint: Assertions Internal〈x〉 and External〈x〉 state whether the object at x
belongs to the module under consideration or not, respectively.

2Runtime configurations include all the information about an execution snapshot: the stack and heap
of frames.

• Change and Authority: An assertion Changes〈x〉 provides conditions for change to

occur; it also called authority.

2.1.3 Bank/Account example

We now put features stated earlier together through the Bank/Account application ob-

tained from object capabilities literature [22]. In particular, we choose the policy Pol_1:

“ With two accounts of the same bank, one can transfer money between them,” and the

policy Pol_2: “Only someone with the bank of a given currency can violate conservation

of that currency.”

Pol_1 states that clients can transfer money between accounts as long as their accounts be-

long to the same bank. It is not a surprise to recognize that it merely sufficient conditions,

expressible within pre-condition and post-condition.

In contrast, the policy Pol_2 captures necessary conditions: Avoiding a bank’s currency

from getting changed without access to the bank is necessary. The policy says that cur-

rency might be increased or decreased by some code if the code involves a function call

in which it is performed by the bank holding the currency. Employing holistic assertions

presented above, we have a holistic specification of Pol_2 as follows.

Pol_2 ≜ b:Bank∧Will
(
Changes

(
b.currency

))
In S

=⇒ ∃o ∈ S. [(o Access b)∧ (o ̸∈ Internal(b))]
(2.1)

Formula 2.1 says that if some execution involving objects, given by the set S, changes the

currency in the bank b at some future time, then there is at least one object from the given

set S that can access the bank b directly, and this object is external to the bank b.

We reformulate Formula 2.1 of Pol_2 into an equivalent one below.

Pol_2 ≜ b:Bank∧ [ ∀o ∈ S. [ ¬(o Access b)∨ (o ∈ Internal(b)) ] ]

=⇒ ¬(
Will

(
Changes

(
b.currency

))
In S

) (2.2)

Formula 2.2 represents that the set S whose elements have direct access to b only if they

are internal to b is insufficient to modify the currency in b at some future time.

Now, let us give some outline to consider how to reason the policy Pol_2. Formally, we

assign:

• P :=∀o ∈ S. [ ¬(o Access b)∨ (o ∈ Internal(b)) ]

• T(o) := [ ¬(o Access b)∨ (o ∈ Internal(b)) ]

• Q := Changes
(
b.currency

)
Formula 2.2 is now equivalent to ∀o ∈ S. T(o) =⇒ ¬(Will(Q) In S). We want to prove

that P is an invariant in the set S. Therefore, we “hope” to establish two separate hypothe-

ses with the first one considered as an invariant as follows.

• (1) ∀o ∈ S. T(o) =⇒ (Next(∀o ∈ S. T(o))) In S

• (2) (∀o ∈ S. T(o) =⇒ ¬ Next(Q)) In S

We “hope” that (1) and (2) are correct, and the conjunction of (1) and (2) that implies

Formula 2.2 is correct as well. So far, to support the above proofs, we also need lemmas

related to spatial connective assertions.

Lemma 1. For any assertions A, B, and a set S, we have

(〈A In S 〉 =⇒ 〈B In S 〉) ≡ 〈(A =⇒ B)In S 〉.

Lemma 2. For any assertions A and a set S, we have

¬〈A In S 〉 ≡ 〈(¬A)In S 〉.

We prove these two technical lemmas in Isabelle in Section 3.7.4 of the next chapter as

well.

2.2 Related Work

The section reviews the foundations and current publications related to the object-

capability model [21]. It also directs on specification and verification for object-capability

programs.

2.2.1 Behavioral Specification Languages

Meyer [20] first presented verification techniques, called “Design by Contract”, for object-

oriented programs, whose specifications along with the form of pre-conditions and post-

conditions on methods, as exemplified by the programming language Eiffel. These ideas

appear in modern specification languages aimed at realistic employment, including Spec#

[1], Dafny [14], JML [13], as well as Whiley [28].

Leino and Schulte [15] used history invariants that have a two-state predicate to specify

the behavior of an object. Their technique was built on an object invariant concept and

was used to verify the invariants of the observer pattern. Another track of work on a

specification for object-oriented programs is Summers and Drossopoulou’s Considerate

Reasoning [29]. They proposed another approach based on object invariants to construct

a specification and verification technique for object-oriented languages.

While these approaches concern specifications on object-oriented languages, they assume

their systems live in a closed world, where other components can be trusted to collaborate.

2.2.2 Object Capabilities and Sandboxes

Miller [21] first introduced the object-capability model, and several recent studies man-

age to verify the correctness or verify the safety of object-capability programs. Google’s

Caja [23], an object-capability subset of Javascript, utilizes sandboxes for securing web

mashups to restrict access of components to ambient authority. Some programming

languages and web systems have utilized the object-capability model, including E [18],

Grace [2], Dart [3], and Wyvern [17]. Miller et al. [21, 24] also define the fundamental

way as defensive consistency “An object is defensively consistent when it can defend its

own invariants and provide correct service to its well-behaved clients, despite arbitrary or

malicious misbehaviors by its other clients.”

Maffeis et al. [16] developed a notation of authority safety based on object capabilities.

They proved that it follows two principles: authority safety is implied from capability

safety and adequate for providing resource isolation. From that, they modeled semantics

for these two principles utilizing small-step operational semantics. They also defined

a language Js, a Javascript subset, to show it is sufficient to provide isolation among

untrusted applications. Moreover, another contribution of the paper pointed out that

these two principles hold for a class of object-capability languages, such as Cajita, a Caja-

based object-capability subset Javascript. However, the Js language does not carry the

object-capability model.

2.2.3 Verification of Object-Capability Programs

There are numerous preliminary studies on the specification and verification of object-

capability programs.

Murray [25] early attempted to formalize defensive consistency and correctness utilizing

process calculi. Murray constructed a process algebra CSP (communicating sequential

processes) model to reason about the security properties of concurrent object-capability

patterns. Then, Murray applied CSP’s model checker FDR to analyze patterns in the

presence of untrusted objects.

Drossopoulou and Noble analyzed Miller’s Mint and Purse [22], a simple example of

capability-based money, and implemented in Joe-E [5] and Grace [27] to argue that cur-

rent specifications are insufficient for reasoning about all aspects of capability policies.

Drossopoulou et al. [6] also proposed specifications for an open world in modeling risk

and trust. They used it to specify an Escrow exchange contract [24], a trusted third agent

that handles exchanges money between untrusted parties. While they focus on the speci-

fication language, they do not define semantics for their predicates, and they only focus

on the example of the Escrow exchange.

A following technical report [7] formalizes and provides the semantics making a connec-

tion for a gap of their early work by proposing a Hoare logic style for reasoning about risk

and trust between objects and formally proving the Escrow protocol meets the specifica-

tion. Although they provide the Hoare logic, their Hoare logic lacks rules for dynamic allo-

cation. Also, there is no proof of the soundness of the Hoare logic. Recently, Drossopoulou

et al. [8] present Chainmail, a specification language for writing holistic specifications,

which concerns both sufficient conditions and necessary conditions.

Devriese et al. [4] adopted a step-indexed Kripke logical relation, as well as a notion

called effect parametricity, to reason about integrity properties of various examples of

capability-wrapped user code in a language with the higher-order state. Furthermore,

they verified some non-trivial examples proving the preservation of invariants on shared

data structures of the user code in the presence of untrusted or unknown code. They also

showed solutions to some example problems, such as a mashup application and the DOM

wrapper. Nevertheless, the model is insufficient in providing no way specifying what an

object-capability pattern does compositionally.

Swasey et al. [30] presented a program logic to specify and verify object-capability pro-

grams compositionally, called OCPL. First, they build their logic based on a framework for

concurrent separation logic, called Iris [10, 12, 11], and mechanized in Coq, a proof assis-

tant. The core idea of OCPL adapts the notion of a low-integrity value in which it can be

shared with untrusted code safely. From that, it identifies the interface between the piece

of verified user and untrusted code. Then, compared to the previous work of Devriese et

al. [4], OCPL gives a modular way of specifying a general property compositionally that

untrusted code can share with used code safely. Finally, they also apply the logic to several

object-capability patterns, including reason about a general membrane, sealer-unsealer

pair, and the caretaker.



Chapter 3

Formalizing Holistic specifications in

Isabelle/HOL

This chapter formalizes most of the core and formal semantics of Chainmail [8] in Is-

abelle/HOL version 2020. It involves formalized definitions, theorems, and technical

lemmas, as well as lemmas supporting Holistic specifications.

Chapter Outline

• Language Syntax. Section 3.1 describes the language syntax of underlying object-

oriented programming.

• Operational Semantics of the Language. Section 3.2 presents the operational se-

mantics of the language mentioned in Section 3.1.

• Module Linking. Section 3.3 covers definitions module linking and proofs of its

properties.

• Module pairs and visible-states semantics. Section 3.4 formalizes the visible-states

semantics and the pairing of an internal module and external module.

• Initial and Arising configurations. Section 3.5 provides the definitions of Initial

and Arising configurations.

• Assertions-Classical Assertions. Section 3.6 gives the formalization of syntax and

semantics of holistic assertions.

• Assertions - Access, Control, Space, Authority, and Viewpoint. Section 3.7 focuses

on the formalization of holistic concepts.

• Summary. Section 3.8 summarizes the formalization of holistic specifications.

We first want to give motivation as to how to formalize the Chainmail in Isabelle/HOL.

Figure 3.1 depicts the method calls rule (methCall_OS) in the operational semantics of

Chainmail, taken from [8]. Given the abstract syntax’s recursive nature, it will not be a

wonder that our pick is an inductive definition. Let us rephrase the method calls rule

(methCall_OS) in natural language. The rule exec_method_call defines the semantics

for method calls of the form x := x0.m(x1, . . . ,xn). It looks up in the current stack frame

φ the object x0 being invoked, producing the address α, which is used to retrieve its class

Class from the heap χ. The method name m of class Class is looked up in the module

M from which the new stack frame φ for the execution of the method is produced. The

continuation cont is updated to remember that a nested call is being executed, whose

result will be assigned to x.

Figure 3.1 Operational semantics of rule methCall_OS of Chainmail [8].

Figure 3.2 Rule exec_method_call presenting in Isabelle for rule methCall_OS.

Following the rules of Figure 3.1, we formalize the rule methCall_OS in Isabelle/HOL,

drawing in Figure 3.2. To formalize it, we define it as an inductive predicate using the

command inductive. The type of exec is Module⇒ Config⇒ Config⇒ bool, where

Module is the set of mappings from class names to class descriptions, and Config is

runtime configurations. We also introduce the concrete syntax (_, _ →e _) for exec.

The Isabelle definition of the method calls rule (exec_method_call) does not directly

connect with the method calls rule (methCall_OS) of Chainmail. The rules make use of

some auxiliary definitions: (1) (ident_lookup φ y) function is used to look up identifier y

in the stack frame φ; (2) (class_lookup χ α) function is used to retrieve the class of the

object whose address is α in the heap χ; (3) build_call_frame is used to create a new

frame in which each value assigns each parameter of method meth in paramValues. The

remaining rules (variable assignment, field assignment, object creation, and return) are

formalized and detailed in Section 3.2.4.

3.1 Language Syntax

In the first part of the formalization, we give the language syntax of underlying object-

oriented programming and its operational semantics and connect other modules to the

module under consideration. We formalize all of them in Isabelle/HOL. Firstly, we are

required to formalize the syntax of the language. We present them by a given set of program

variables Identifier, a set of field names FieldName, a set of class names ClassName, and

a set of method names MethodName. The fields of the language declared are untyped, the

method consists of an untyped parameter, with no return type as well, because the core

language of Holistic specifications is untyped.

type_synonym Identifier = nat

typedecl FieldName

typedecl ClassName

typedecl MethodName

Statements Stmt in the programming language consist of field read ReadFromField, field

write AssignToField, method call MethodCall, object creation NewObject, and return state-

ment Return.

datatype Stmt = AssignToField FieldName Identifier

| ReadFromField Identifier FieldName

| MethodCall Identifier Identifier MethodName "Identifier list"

| NewObject Identifier ClassName "Identifier list"

| Return Identifier

The sequences of a statement are declared as Stmts as well. It consists of a single statement

and a sequence of statements.

datatype Stmts = SingleStmt Stmt | Seq Stmt Stmts

The method body consists of sequences of statements and their parameters, which are a

list of identifiers. Notice that we store the method names inside the class to make lookup

easier.

record Method = formalParams :: "Identifier list"

body :: Stmts

The class description consists of field and method declarations. Note that we omit the

ClassId as it appears to be duplicated in the Module. We also assume that every method is

defined on every object for the sake of simplicity.

record Class = objFields :: "FieldName list"

methods :: "MethodName ⇒ Method"

A module is defined as the set of mappings from class names to class descriptions. Note

that the class name is distinct from local variables. Every class is not defined by every

module, as otherwise linking becomes meaningless.

type_synonym Module = "ClassName ⇒ Class option"

Method lookup function M returns a method m for a class C inside the module M.

definition

M :: "Module ⇒ ClassName ⇒ MethodName ⇒ Method option"

where

"M M C m ≡ case (M C) of None ⇒ None | Some c ⇒ Some ((methods c) m)"

3.2 Operational Semantics of the Language

3.2.1 Interpretations

Heaps χ are mappings from addresses α to objects obj. Stack frames ϕ are mappings

from identifiers x plus the distinguished identifier this to values, where values include

addresses. Stack frames also store the current continuation (code to be executed).

Configurations σ are pairs (ψ, χ) where ψ is a list of stack frames ϕ and χ is the heap.

The following notation we use throughout the upcoming sections.

• Lookup of fields f on object with address α in the heap χ, field_lookup χ α f, is

written χ(α, f).

• The class of the object whose address is α in heap χ, class_lookup χ α, is written

Class(α)χ.

• Lookup of the class of the this object in the runtime configurations σ,

this_class_lookup σ, is written Class(this)σ.

• Lookup of identifier x in the frame ϕ, ident_lookup ϕ x, is written ⌊x ⌋ϕ.

• Lookup of identifier x in context σ, evalVar x σ, is written ⌊x ⌋σ.

• Lookup of field f from this object in frame ϕ and heap χ, this_field_lookup ϕ χ

f, is written ⌊this.f ⌋(ϕ,χ).

• Update the variable map of frame ϕ so that variable x maps to value v,

frame_ident_update ϕ x v, is written ϕ[x 7→ v].

• Update the object at address α on the heap χ to the object obj, heap_update χ α

obj, is written χ[α 7→ obj].

• To obtain the continuation cont from the frame ϕ, we write cont ϕ.

3.2.2 Runtime Entities

We are ready to introduce the addresses Addr as an enumerable set and null Null.

type_synonym Addr = nat

consts Null :: Addr

Then, we also define values Values, consisting of null, addresses, and sets of addresses

VAddrSet.

datatype Value = VAddr Addr | VAddrSet "Addr set"

Continuations are either statements or a nested call followed by statements. Frames

consist of a continuation, a mapping from identifiers to values, and an address this.

datatype Continuation = Code Stmts | NestedCall Identifier Stmts

record Frame = cont :: Continuation

vars :: "Identifier ⇒ Value option"

this :: Addr

Stacks are sequences of frames. Objects consist of a class identifier className and a map-

ping from field name FieldName to values values. Heaps Heap are defined as mappings

from addresses to objects. Lastly, runtime configurations Config are pairs of stacks and

heaps.

We use the option type to represent partial functions. i.e., the partial function from a to b

is represented by the type a ⇒ b option. For instance, the objFields is a partial function

from FieldName to Value defined below.

type_synonym Stack = "Frame list"

record Object = className :: ClassName

objFields :: "FieldName ⇒ Value option"

type_synonym Heap = "Addr ⇒ Object option"

type_synonym Config = "Stack × Heap"

3.2.3 Lookup and update of runtime configurations

We represent interpretations of a field lookup and a class lookup. The field_lookup

function is used to retrieve the value stored in field f of the object at address α in the heap

χ. It is a partial function since there might be no such object at address α.

fun

field_lookup :: "Heap ⇒ Addr ⇒ FieldName ⇒ Value option"

where

"field_lookup χ α f =

(case (χ α) of None ⇒ None

| Some obj ⇒ (objFields obj) f)"

The function class_lookup is used to retrieve the class of the object obj whose address is

α in the heap χ. It is a partial function since there might be no such class at address α.

fun

class_lookup :: "Heap ⇒ Addr ⇒ ClassName option"

where

"class_lookup χ α =

(case (χ α) of None ⇒ None

| Some obj ⇒ Some (className obj))"

The function ident_lookup is used to retrieve identifier x in the frame ϕ. It is a partial

function since there might be no such identifier in the frame ϕ.

fun

ident_lookup :: "Frame ⇒ Identifier ⇒ Value option"

where

"ident_lookup ϕ x = vars ϕ x"

The function this_field_lookup is used to retrieve the class of the this object in the

runtime configuration σ. It is a partial function since there might be no such class in the

runtime configuration σ.

fun

this_field_lookup :: "Frame ⇒ Heap ⇒ FieldName ⇒ Value option"

where

"this_field_lookup ϕ χ f =

(case χ (this ϕ) of None ⇒ None

| Some obj ⇒ objFields obj f)"

The auxiliary function this_field_update is used to update the field f of the distinguished

this object to refer to value v, given the stack frame ϕ and heap χ.

fun

this_field_update :: "Frame ⇒ Heap ⇒ FieldName ⇒ Value ⇒ Heap option"

where

"this_field_update ϕ χ f v =

(case χ (this ϕ) of None ⇒ None

| Some obj ⇒ Some (χ(this ϕ :=

Some (obj (|objFields := ((objFields obj)(f := Some v)) |)))))"

The function frame_ident_update updates the variable map of the stack frame ϕ so that

variable x maps to value v.

fun

frame_ident_update :: "Frame ⇒ Identifier ⇒ Value ⇒ Frame"

where

"frame_ident_update ϕ x v = (ϕ(|vars := ((vars ϕ)(x := Some v)) |))"

The function heap_update updates the object at address α on the heap χ to map to the

object obj.

fun

heap_update :: "Heap ⇒ Addr ⇒ Object ⇒ Heap"

where

"heap_update χ α obj = χ (α := Some obj)"

The function this_class_lookup is used to look up the class of the this object in the

runtime configuration σ.

fun

this_class_lookup :: "Config ⇒ ClassName option"

where

"this_class_lookup σ =

(case σ of (ϕ#ψ,χ) ⇒
(case χ (this ϕ) of None ⇒ None

| Some obj ⇒ Some (className obj))

| _ ⇒ None)"

Note that these above functions return option in case of lookup failure.

3.2.4 Operational semantics

Now, we turn to the operational semantics of the programming language. We define it as

an inductive predicate called exec. Its rules make use of the following auxiliary definitions.

The auxiliary function build_call_frame is used to create a new frame in which the this

object is assigned by addressα, and each value also assigns each parameter of method meth

in paramValues. We also need a condition that the length of method meth equals the length

of the list paramValues. We implement such a condition in the rule exec_method_call.

definition

build_call_frame :: "Method ⇒ Addr ⇒ Value list ⇒ Frame"

where

"build_call_frame meth α paramValues ≡
(|cont = Code (body meth),

vars = map_of (zip (formalParams meth) paramValues),

this = α|)"

The auxiliary function build_new_object creates a new object in which each value in

the list fieldValues assigns each field in class c. Such a function is useful for the object

creation’s rule exec_new. Same as the function build_call_frame, the length of both the

list of fields in class c and the list fieldValues must equal.

definition

build_new_object :: "ClassName ⇒ Class ⇒ Value list ⇒ Object"

where

"build_new_object C c fieldValues ≡
(|className = C,

objFields = map_of (zip (Class.objFields c) fieldValues) |)"

In object creation’s rule exec_new, we need a fresh address in heap χ, i.e., the new address

utterly different from addresses including the current heap χ. The function fresh_nat

generates such a fresh address.

definition

fresh_nat :: "Identifier set ⇒ Identifier"

where

"fresh_nat X = (if X = {} then 0 else (Suc (last (sorted_list_of_set X))))"

Lemma fresh_nat_is_fresh says that such a fresh address is not in the current heap χ.

lemma fresh_nat_is_fresh [simp]:

"finite X =⇒ fresh_nat X ∉ X"

apply (induct rule: finite.induct)

apply simp

apply(clarsimp simp: fresh_nat_def)

using not_eq_a not_in_A by auto

An operational semantics is represented as a set of rules given formally below. These

rules are method calls exec_method_call, variable assignment exec_var_assign, field

assignment exec_field_assign, object creation exec_new, and return exec_return.

The rule exec_method_call defines the semantics for method calls of the form x :=

y.m(params). It looks up in the current stack frame ϕ the object y being invoked, pro-

ducing the address α, which is used to retrieve its class C from the heap χ; the rule also

looks up each identifier in params, checking to make sure that each such lookup succeeds.

The method name m of class C is looked up in the module M from which the new stack

frame ϕ’’ for the execution of the method is produced. The continuation is updated to

remember that a nested call is being executed, whose result will be assigned to x.

The rule exec_var_assign defines the semantics for field read of form x := this.y. The

continuation is updated to remember that a sequence of statements stmts is being exe-

cuted. The result of the lookup of the field y from this object in the frame ϕ and the heap

χ will be assigned to x.

The rule exec_field_assign defines the semantics for field write of form this.y = x.

The continuation is updated to remember that a sequence of statements stmts is being

executed. The rule looks up identifier x in the current stack frame ϕ, producing the value

v, which is updated to the field y to create a new heap χ’ given the stack frame ϕ and

heap χ.

The rule exec_new defines the semantics for object creation of form

x := new C(params). The rule looks up each identifier in params, checking to make sure

that each such lookup succeeds. A fresh address α will update the evaluation of identifier

x in the current heap σ. The heap χ at the address α will be updated by a new object,

which each field declared in the method m of class C will be updated by each value of each

identifier in params, respectively.

The rule exec_return defines the semantics for the return statement of form return x.

The value of identifier x in stack frame σ is assigned to the identifier x’ where the result

of a nested call is assigned. The continuation is updated to remember that a statement

stmts’ followed by the nested call is being executed.

inductive

exec :: "Module ⇒ Config ⇒ Config ⇒ bool" ("_, _ →e _")

where

exec_method_call:

"cont ϕ = Code (Seq (MethodCall x y m params) stmts) =⇒
ident_lookup ϕ y = Some (VAddr α) =⇒
class_lookup χ α = Some C =⇒
paramValues = map (ident_lookup ϕ) params =⇒
None ∉ set paramValues =⇒
M M C m = Some meth =⇒
length (formalParams meth) = length params =⇒
ϕ’’ = build_call_frame meth α (map the paramValues) =⇒
M, (ϕ # ψ, χ) →e (ϕ’’ # (ϕ(|cont := NestedCall x stmts |)) # ψ, χ)" |

exec_var_assign:

"cont ϕ = Code (Seq (ReadFromField x y) stmts) =⇒



3.2 Operational Semantics of the Language 25

M, (ϕ # ψ, χ) →e
((ϕ (| cont := Code stmts ,

vars := ((vars ϕ)(x := this_field_lookup ϕ χ y)) |) ) # ψ, χ)" |

exec_field_assign:

"cont ϕ = Code (Seq (AssignToField y x) stmts) =⇒
ident_lookup ϕ x = Some v =⇒
this_field_update ϕ χ y v = Some χ’ =⇒
M, (ϕ # ψ, χ) →e (ϕ (|cont := Code stmts |) # ψ, χ’)" |

exec_new:

"cont ϕ = Code (Seq (NewObject x C params) stmts) =⇒
paramValues = map (ident_lookup ϕ) params =⇒
None ∉ set paramValues =⇒
M C = Some c =⇒
length params = length (Class.objFields c) =⇒
obj’ = build_new_object C c (map the paramValues) =⇒
α = fresh_nat (dom χ) =⇒
χ’ = χ(α := Some obj’) =⇒
M, (ϕ # ψ, χ) →e (ϕ (|cont := Code stmts,

vars := ((vars ϕ)(x := Some (VAddr α))) |) # ψ, χ’)" |

exec_return:

"cont ϕ = Code (SingleStmt (Return x)) ∨
cont ϕ = Code (Seq (Return x) stmts) =⇒
cont ϕ’ = NestedCall x’ stmts’ =⇒
M, (ϕ # ϕ’ # ψ, χ) →e ((ϕ’ (| cont := Code stmts’,

vars := ((vars ϕ) (x’ := ident_lookup ϕ x)) |)) # ψ, χ )"

We formally define the execution of more steps exec_rtrancl, which is the reflexive,

transitive closure of exec, as follows.

inductive

exec_rtrancl:: "Module ⇒ Config ⇒ Config ⇒ bool" ("_, _ →e* _")

where

exec_rtrancl_equiv: "σ = σ’ =⇒ M, σ →e* σ’" |

exec_rtrancl_trans: " [[ M, σ →e* σ’’; M, σ’’ →e σ’ ]] =⇒ M, σ →e* σ’"

To have connections with Chainmail, we introduce the concrete syntax (_, _ →e _) for

exec and (_, _ →e* _) for exec_rtrancl.

3.3 Module Linking

In an open world, to reason about the operation of a module, we need to talk about how

it behaves when operating in the presence of other (possibly untrusted) code. For that

purpose, we define what it means to combine two modules in a module linking operator.

Later, this will be used to model the operation of module M in the presence of other

untrusted modules M’ that it is linked to.

We know that the linking should be well-formed. So, the function link_wf represents the

well-formedness of the linking, saying that when the two modules do not both define the

same class (i.e., when their domains are disjoint).

definition

link_wf :: "Module ⇒ Module ⇒ bool"

where

"link_wf M M’ ≡ dom M ∩ dom M’ = {}"

definition

moduleAux :: "Module ⇒ Module ⇒ ClassName ⇒ Class option" ( infixl "◦aux" 55)

where

"(M ◦aux M’) ≡ λC. (if C ∈ dom M then M C else M’ C)"

The next step is to define a module linking operator. We introduce concrete syntax ◦l
for the module linking operator moduleLinking. The function moduleLinking takes two

modules M and M’ and returns the union of the two.

definition

moduleLinking :: "Module ⇒ Module ⇒ Module" ( infix "◦l" 55)

where

"M ◦l M’ ≡ (M ◦aux M’)"

When the linking is well-formed, it should be commutative and associative. We prove that

linking is commutative (link_commute ) and associative (link_assoc ) when well-formed.

lemma link_commute [simp]: "link_wf M M’ =⇒ M ◦l M’ = M’ ◦l M"

unfolding moduleLinking_def moduleAux_def dom_def

apply (simp cong: if_cong)+

apply (auto simp: link_wf_def)

by fastforce

lemma link_assoc [simp]:

"link_wf M M’ =⇒ (M ◦l M’) ◦l M’’ = M ◦l (M’◦l M’’)"

unfolding moduleLinking_def moduleAux_def dom_def link_wf_def

apply (simp cong: if_cong)+

by auto

3.4 Module pairs and visible-states semantics

Holistic specifications are useful to talk about the interactions between a module M and

other potentially untrustworthy modules M’ that it might interact with. We formally

capture these interactions via visible-state semantics in which the visible states are

those as seen from outside the module M, i.e., those in which some M’ object is running.

We name the module M and M’ for the internal module and external module, respectively.

This section introduces the formalization of module pairs and visible-states semantics. A

visible execution is a sequence of execution steps that looks like this: σ →e σ2 →e ...

σn-1 →e σn where the class of the this object in σ comes from the external module M’

and the class of the this object in σn also comes from the external module M’. However,

the class of the this object in every other σ2, ..., σ(n-1) comes from the internal module M.

We capture this using two inductive definitions. The first one, defined as internal_exec,

talks about the first n - 2 steps of execution, each such step leads to a configuration σi ,

where 2 ≤ i < n in which the this object of σi is in the internal module M.

inductive

internal_exec ::

"Module ⇒ Module ⇒ Config ⇒ (Config list) ⇒ Config ⇒ bool"

("_;_,_ →ei 〈_ 〉 _") for M :: Module and M’ :: Module

where

internal_exec_first_step:

"link_wf M M’ =⇒
(M ◦l M’), σ →e σ’ =⇒
this_class_lookup σ = Some c =⇒ c ∈ dom M’ =⇒
this_class_lookup σ’ = Some c’ =⇒ c’ ∈ dom M =⇒
internal_exec M M’ σ [σ’] σ’" |

internal_exec_more_steps:

"internal_exec M M’ σ tr σ’ =⇒
(M ◦l M’), σ’ →e σ’’ =⇒

this_class_lookup σ’’ = Some c =⇒ c ∈ dom M =⇒
internal_exec M M’ σ (tr@[σ’’]) σ’’"

The second inductive definition as visible_exec is just for the final step from σ(n-1) to

σn .

inductive

visible_exec :: "Module ⇒ Module ⇒ Config ⇒ Config ⇒ bool" ("_;_,_ →e _")

where

visible_exec_intro:

"internal_exec M M’ σ tr σ’ =⇒
(M ◦l M’), σ’ →e σ’’ =⇒
this_class_lookup σ’’ = Some c =⇒
c ∈ dom M’ =⇒
visible_exec M M’ σ σ’’"

3.4.1 Determinism

There is a critical thing that we need to prove in the execution of the language, and the exe-

cution of the visible-state semantics is deterministic. The language is deterministic when

any two executions start from the same state step to the same next state. Formally, the

execution of the language or the visible-states is deterministic if for the same initial state σ,

there is at most one next state that is reached after one step of execution, i.e., if σ steps to

σ’ and also to σ’’, then σ’ = σ’’. The lemma exec_det shows that the execution of the

language is deterministic. The lemma visible_exec_det also shows that the execution of

the visible-state semantics is deterministic. To prove this lemma, we need several technical

definitions and sub-lemmas such as internal_exec_rev’, internal_exec_is_internal,

or internal_exec_appD (See Appendix A.2).

To prove the determinism of visible-states semantics, we first need to prove that the

internal_exec definition is deterministic. To do that, it helps to have an equivalent

definition of it that operates in reverse. That definition we call internal_exec_rev, which

is defined as follows via the intermediate definition internal_exec_rev’.

inductive

internal_exec_rev’ ::

"Module ⇒ Module ⇒ Config ⇒ (Config list) ⇒ Config ⇒ bool"

("_;_,_ →eir1 〈_ 〉 _") for M :: Module and M’ :: Module

where

internal_refl:

"internal_exec_rev’ M M’ σ [] σ" |

internal_step:

"(M ◦l M’), σ →e σ’ =⇒
this_class_lookup σ = Some c =⇒ c ∈ dom M =⇒
this_class_lookup σ’ = Some c’ =⇒ c’ ∈ dom M =⇒
internal_exec_rev’ M M’ σ’ tr σ’’ =⇒
internal_exec_rev’ M M’ σ (σ’#tr) σ’’"

inductive

internal_exec_rev ::

"Module ⇒ Module ⇒ Config ⇒ (Config list) ⇒ Config ⇒ bool"

("_;_,_ →eir 〈_ 〉 _") for M :: Module and M’ :: Module

where

internal_exec_rev_first_step:

"link_wf M M’ =⇒
(M ◦l M’), σ →e σ’ =⇒
this_class_lookup σ = Some c =⇒ c ∈ dom M’ =⇒
this_class_lookup σ’ = Some c’ =⇒ c’ ∈ dom M =⇒
internal_exec_rev’ M M’ σ’ tr σ’’ =⇒
internal_exec_rev M M’ σ (σ’#tr) σ’’"

Finally, we conclude that internal_exec_det is deterministic. It is one of the main lemmas

needed to show that the visible-states semantics is deterministic.

Lemma internal_exec_det says that in the first n - 2 steps of execution if for the same

initial state σ, there is at most one next state that is reached after one step of execution,

i.e., if σ steps to σ’ and also to v, then v = σ’.

lemma internal_exec_det:

"M;M’, σ →ei 〈tr 〉 σ’ =⇒ M;M’, σ →ei 〈tr 〉 v =⇒ v = σ’"

by (auto simp: internal_exec_det_aux)

Lemma visible_exec_det asserts that the execution of the visible-states semantics is

deterministic as below. Similar to Lemma internal_exec_det, Lemma visible_exec_det

also says that if σ steps to σ’ and also to σ’’, then σ’’ = σ’, but in this case, it is the final

step from σ(n-1) to σn .

lemma visible_exec_det:

"M;M’, σ →e σ’ =⇒ link_wf M M’ =⇒ M;M’, σ →e σ’’ =⇒ σ’’ = σ’"

by (auto simp: visible_exec_det_aux)

The lemma exec_det below asserts that the execution of the languages is deterministic.

The proof is by structural induction on the definition of exec.

lemma exec_det:

"M, σ →e σ’ =⇒ M, σ →e σ’’ =⇒ σ’’ = σ’"

by (auto simp: exec_det_aux)

We make the proof details of the determinism of language and visible-states at the lemma

exec_det_aux and lemma visible_exec_det_aux, respectively, in Appendix A.2.

3.4.2 Linking modules preserving execution

Intuitively, taking a module M and placing it in a larger context M’ cannot reduce the

behaviors of M. Therefore, if M can perform some execution step on its own, we would

expect it also to perform that same step when linked against an arbitrary module M’. We

formally prove this below.

A similar argument also applies to the visible state semantics. If M when linked against M’

has a visible execution, it should still have that same execution when linked against M’ ◦l
M’’. We formally prove this property also.

Together these properties tell us that linking is monotonic for a module’s executions (i.e.,

increasing the context increases the possible executions but does not remove any), as

should be expected.

In this section, we place proofs of module linking preserving execution. First, we need to

define link_wf_3M for three modules whose domains are pairwise disjoint.

definition

link_wf_3M :: "Module ⇒ Module ⇒ Module ⇒ bool"

where

"link_wf_3M M M’ M’’ ≡ ((dom M ∩ dom M’ = {}) ∧
(dom M’ ∩ dom M’’ = {}) ∧
(dom M’’ ∩ dom M = {}))"

Technical lemma link_dom says that the union of two domains of two modules M and M’

is the two’s union domain.

lemma link_dom [simp]:

"dom (M ◦l M’) = dom M ∪ dom M’"

by (auto simp: moduleLinking_def moduleAux_def dom_def)

We also need technical lemma link_wf_3M_dest, saying that if three modules whose do-

mains are pairwise disjoint are well-formed, pair arbitrary modules are also well-formed.

lemma link_wf_3M_dest [simp,intro,dest]:

"link_wf_3M M M’ M’’ =⇒ link_wf M M’"

"link_wf_3M M M’ M’’ =⇒ link_wf M’ M’’"

"link_wf_3M M M’ M’’ =⇒ link_wf M M’’"

"link_wf_3M M M’ M’’ =⇒ link_wf (M ◦l M’) M’’"

by(fastforce simp: link_wf_def link_wf_3M_def)+

Then, we put the lemma link_exec to show that the module linking preserves one-module

if its module linking is defined.

lemma link_exec:

" [[ M, σ →e σ’; link_wf M M’ ]] =⇒ (M ◦l M’), σ →e σ’"

by (simp add: link_exec_aux)

Moreover, the module linking preserves visible state semantics also, as shown in lemmas

visible_exec_linking_1 and visible_exec_linking_2. Since the definition of visible-

state semantics visible_exec, is based on the definitions of internal execution

internal_exec, we need to prove that the internal execution is also preserved by linking

as shown in lemmas internal_linking_1 and internal_linking_2 as well.

Proofs of lemma internal_linking_1 and internal_linking_2 are by structural induc-

tion on the definition of internal_exec.

The lemma internal_linking_1 guarantees that M; M’,σ →ei 〈tr 〉 σ’ implies that all

intermediate configurations are external to M’ and thus also to M’ ◦l M’’.

lemma internal_linking_1:

" [[M; M’,σ →ei 〈tr 〉 σ’; link_wf_3M M M’ M’’ ]] =⇒
M;(M’ ◦l M’’), σ →ei 〈tr 〉 σ’"

by (simp add: internal_linking_1_aux)

Similarly, the lemma internal_linking_2 guarantees that M; M’,σ →ei 〈tr 〉 σ’ implies

that all intermediate configurations are internal to M and thus also to M ◦l M’’.

lemma internal_linking_2:

" [[ M; M’,σ →ei 〈tr 〉 σ’ ; link_wf_3M M M’ M’’ ]] =⇒
(M ◦l M’’); M’, σ →ei 〈tr 〉 σ’"

by (simp add: internal_linking_2_aux)

Thanks to two useful lemmas internal_linking_1 and internal_linking_1, we also gain

the guarantee of module linking preserves visible state semantics. Proofs of lemmas

visible_exec_linking_1 and visible_exec_linking_2 are by structural induction on

the definition of visible_exec.

lemma visible_exec_linking_1:

" [[(M;M’,σ →e σ’); (link_wf_3M M M’ M’’) ]] =⇒
M; (M’ ◦l M’’), σ →e σ’"

by (simp add: visible_exec_linking_1_aux)

lemma visible_exec_linking_2:

" [[(M;M’,σ →e σ’); (link_wf_3M M M’ M’’) ]] =⇒
(M ◦l M’’); M’, σ →e σ’"

by (simp add: visible_exec_linking_2_aux)

We make the proof details of Linking modules preserving execution in Appendix A.3.

3.5 Initial and Arising configurations

What does it mean for a holistic specification to hold for a module M when linked against

some external module M’ ? It means that the property holds for all Arising configurations

of M with M’. These are the configurations that can be reached in the visible state semantics

of M with M’ when execution begins from the initial, empty configuration. We formally

define these ideas below.

Now, we assume that the initial stack frame maps no local variables. Note that we let the

continuation be arbitrary.

definition

initial_frame :: "Frame ⇒ bool"

where

"initial_frame ϕ ≡ (vars ϕ = Map.empty ∧ this ϕ = Null)"

Suppose we have defined the execution of more steps exec_rtrancl, which is the reflexive,

transitive closure of exec. In that case, we also want to define the execution of more steps

exec_module, saying that it is the reflexive, transitive closure of visible execution exec. We

formally define the execution exec_module as follows.

inductive

exec_module ::

"Module ⇒ Module ⇒ Config ⇒ Config ⇒ bool" ("_;_, _ →e* _")

where

exec_module_equiv: "σ = σ’ =⇒ M;M’, σ →e* σ’" |

exec_module_trans: " [[(M;M’, σ →e* σ’’); (M; M’, σ’’ →e σ’) ]] =⇒
M;M’, σ →e* σ’"

Initial configurations Initial might contain arbitrary code but no objects.

definition

Initial :: "(Stack × Heap) ⇒ bool"

where

"Initial σ ≡ (case σ of (ψ,χ) ⇒
(case ψ of ([ϕ]) ⇒

initial_frame ϕ ∧ χ = Map.empty

| _ ⇒ False))"

From initial configurations Initial, execution of code from module-pair (M; M’), creates

a set of Arising configurations Arising.

definition

Arising :: "Module ⇒ Module ⇒ Config set "

where

"Arising M M’ ≡ {σ.∃σ0. (Initial σ0 ∧ (M;M’, σ0 →e* σ))}"

Notice that M;M’, σ0 →e* σ is visible-state semantics introduced in Section 3.4.

3.6 Assertions - Classical Assertions

We have defined the object-based programming language and its semantics, including the

visible state semantics, and proved them deterministic. However, we have not yet defined

the language in which holistic specifications are expressed. We now do that by formally

defining the assertions of holistic specifications and giving them meaning over the visible

state semantics of the programming language defined above. We give the formalization of

syntax and semantics of holistic assertions in this section.

3.6.1 Syntax of Assertions and its standard semantics

The validity of assertions Assertion has a form of M; M’,σ |= A where the module M

and M’ are internal and external, respectively. The assertion returns a bool option type

rather than a bool. For instance, if we compare two expressions e and e’ and one of

them evaluates to None, then the semantics of the comparison is undefined. Hence the

semantics of assertions is partial, represented using the option type as with other partial

functions.

Unlike the syntax of the programming language, which is deeply embedded, we decided

to embed assertions in our formalization shallowly. It was done to enable us to extend the

set of assertions, later on, more efficiently.

type_synonym Assertion = "Module ⇒ Module ⇒ Config ⇒ bool option"

Assertions consist of pure expressions such as atrue and afalse.

datatype Expr = ENull | EId Identifier | EField Expr FieldName

definition

atrue :: "Assertion"

where

"atrue ≡ λM M’ σ. Some True"

definition

afalse :: "Assertion"

where

"afalse M M’ σ ≡ Some False"

Expressions support nested field lookups, e.g., x.f.g via (EField (EField (EId x) f)

g).

fun

evalVar :: " Identifier ⇒ Config ⇒ Value option"

where

"evalVar x (ϕ#ψ,χ) = ident_lookup ϕ x" |

"evalVar x ([],χ) = None"

Recall that expressions denote values. We, therefore, define the semantics of expressions

via the following partial function. Note that expression might not evaluate a value, e.g.,

for a field lookup for a non-existent object, in which case the semantics returns None.

Otherwise, it returns Some v, where v is the value the expression denotes, in configuration

σ.

primrec

expr_eval :: "Expr ⇒ Config ⇒ Value option"

where

"expr_eval ENull σ = Some (VAddr Null)" |

"expr_eval (EId x) σ = evalVar x σ" |

"expr_eval (EField e f) σ =

(case (expr_eval e σ) of Some (VAddr a) ⇒
field_lookup (snd σ) a f)"

We define generic comparisons between expressions. For example, the notation of greater

than would be expressed as acompare (>) e e’.

definition

acompare :: "(Value ⇒ Value ⇒ bool) ⇒ Expr ⇒ Expr ⇒ Assertion"

where

"acompare c e e’ ≡ λM M’ σ.

(case (expr_eval e σ) of Some v ⇒
(case (expr_eval e’ σ) of Some v’ ⇒ Some (c v v’)

| None ⇒ None)

| None ⇒ None)"

We give formalized definitions of the semantics of assertions involving expressions. The

partial function expr_class_lookup is used to look up the class where expression e is

located in the runtime configuration σ.

fun

expr_class_lookup :: "Config ⇒ Expr ⇒ ClassName option"

where

"expr_class_lookup σ e =

(case σ of (ϕ#ψ,χ) ⇒
(case (expr_eval e σ) of Some (VAddr a) ⇒

(case χ a of Some obj ⇒ Some (className obj) | None ⇒ None ))

| _ ⇒ None)"

The assertion aExpClassId states whether an expression e belongs to a class identifier

ClassId.

definition

aExpClassId :: "Expr ⇒ ClassName ⇒ Assertion"

where

"aExpClassId e ClassId ≡
λM M’ σ. (case (expr_class_lookup σ e) of None ⇒ None

| Some cid ⇒ Some (cid = ClassId))"

The function expInS checks the address of the expression e is in the set of addresses of

the given set S.

fun

expInS :: "Config ⇒ Expr ⇒ Identifier ⇒ bool option"

where

"expInS σ e S =

(case (expr_eval e σ) of Some (VAddr a) ⇒
(case (evalVar S σ) of Some v ⇒

(case v of VAddr addr ⇒ None |

VAddrSet addrSet ⇒ (Some (a ∈ addrSet))) |

None ⇒ None) |

None ⇒ None)"

The assertion aExpInS presents whether an expression e belongs to a given set S.

definition

aExpInS :: "Expr ⇒ Identifier ⇒ Assertion"

where

"aExpInS e S ≡ λ M M’ σ. (expInS σ e S)"

We formalize the meaning of standard logical connectives between assertions. For exam-

ple, the logical conjunction of two assertions A and A’ is expressed as aAnd A A’. Similarly,

the logical disjunction, negation, and implication are expressed as aOr A A’, aNot A, and

aImp A A’, respectively.

To support such assertions, we define a generic binary operator between assertions. For

example, the notation of implication would be expressed as bopt (−→) (A M M’ σ) (A’

M M’ σ).

definition

bopt ::

"(bool ⇒ bool ⇒ bool) ⇒ bool option ⇒ bool option ⇒ bool option"

where

"bopt f a b ≡
(case a of Some a’ ⇒

(case b of Some b’ ⇒ Some (f a’ b’)

| None ⇒ None)

| None ⇒ None)"

definition

aImp :: "Assertion ⇒ Assertion ⇒ Assertion"

where

"aImp A A’ ≡ λM M’ σ. bopt (−→) (A M M’ σ) (A’ M M’ σ)"

definition

aAnd:: "Assertion ⇒ Assertion ⇒ Assertion"

where

"aAnd A A’ ≡ λM M’ σ. bopt (∧) (A M M’ σ) (A’ M M’ σ)"

definition

aOr:: "Assertion ⇒ Assertion ⇒ Assertion"

where

"aOr A A’ ≡ λM M’ σ. bopt (∨) (A M M’ σ) (A’ M M’ σ)"

definition

aNot:: "Assertion ⇒ Assertion"

where

"aNot A ≡λM M’ σ. case (A M M’ σ) of None ⇒ None | Some a’ ⇒ Some (¬ a’)"

We also give the universal and existential quantification for holistic assertions. The univer-

sal quantification is expressed formally as aAll fA, and the existential quantification is

presented as aEx fA. We represent an assertion like ∀x. P x, by having P be a function

that takes the identifier x as an argument and returns an assertion. It is an instance of

Higher-Order Abstract Syntax.

definition

aAll :: "(Identifier ⇒ Assertion) ⇒ Assertion"

where

"aAll fA ≡ λM M’ σ. (if (∃v’. fA v’ M M’ σ = None)

then None

else Some (∀v. the (fA v M M’ σ)))"

It is similar to an assertion like ∃x. P x.

definition

aEx :: "(Identifier ⇒ Assertion) ⇒ Assertion"

where

"aEx fA ≡ λM M’ σ. (if (∃v’. fA v’ M M’ σ = None)

then None

else Some (∃v. the (fA v M M’ σ)))"

3.6.2 Properties of classical logic

Here, we deliver formal proofs of properties that apply to conjunction, disjunction, nega-

tion, implication, universal quantification, as well as existential quantification to show

that holistic assertions are classical. Remember that assertions are partial. These proper-

ties hold only for well-formed assertions whose semantics are not undefined. We capture

this formally via the definition Assert_wf below.

definition

Assert_wf:: "Assertion ⇒ Module ⇒ Module ⇒ Config ⇒ bool"

where

"Assert_wf A M M’ σ ≡ A M M’ σ ̸= None"

We are taking an example to show that holistic assertions are distributive property

aDistributive_1 of logical conjunction over logical disjunction for assertion A, A’ and

A’’. Other properties can be found in Section A.5.

lemma aDistributive_1:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ; Assert_wf A’’ M M’ σ ]] =⇒
(aAnd (aOr A A’) A’’ M M’ σ) = (aOr (aAnd A A’’) (aAnd A’ A’’) M M’ σ)"

unfolding aAnd_def aOr_def bopt_def option.case_eq_if by auto

3.7 Assertions - Access, Control, Space, Authority, and View-

point

In this section, we focus on the formalization of holistic concepts. These consist of

permission, control, space, authority, and viewpoint.

3.7.1 Access

Access or permission states an object has a direct path to another object. In more detail,

in the current frame, the access assertions are defined as three cases: (1) two objects are

aliases, (2) the first points to an object with a field whose value is the same as the second

object, (3) the first object is currently executing an object, and the second object is a local

parameter that appears in the code in the continuation.

In particular, we formalize access assertion Access with supporting functions thisEval

and evalThis. The partial function thisEval is used to look up the address of this object

in runtime configuration σ. The function returns None in case of lookup failure.

fun

thisEval :: "Config ⇒ Value option"

where

"thisEval σ =

(case σ of (ϕ#_, _) ⇒
(case (this ϕ) of

addr ⇒ Some (VAddr addr))

| _ ⇒ None)"

The partial function evalThis is used to look up the field f from this object in runtime

configuration σ. The function returns None in case of lookup failure.

fun

evalThis :: "FieldName ⇒ Config ⇒ Value option"

where

"evalThis f σ =

(case σ of (ψ,χ) ⇒
(case ψ of [] ⇒ None

| (ϕ#ψ) ⇒ this_field_lookup ϕ χ f))"

The function ConfigCont obtains the continuation cont from the configuration σ.

definition

ConfigCont :: "Config ⇒ Continuation"

where

"ConfigCont σ ≡ (case σ of (ϕ#_, _) ⇒ cont ϕ)"

The assertion Access x y holds if in runtime configuration σ,

(1) the value of identifier x and y is the same, or

(2) there exists a field f such that the value of the field f and the identifier y is the same, or

(3) the value of identifier x and the this object is the same, as well as the value of identifier

y and z is also the same, where z is a local parameter in continuation cont.

definition

Access:: "Identifier ⇒ Identifier ⇒ Assertion"

where

"Access x y ≡ λM M’ σ.

if (evalVar x σ = None ∨ evalVar y σ = None)

then None

else Some ( (evalVar x σ = evalVar y σ) ∨
(∃f. (evalThis f σ = evalVar y σ)) ∨
((evalVar x σ = thisEval σ) ∧
(∃z z1 stmts. ((Code stmts = ConfigCont σ) ∨
(NestedCall z1 stmts = ConfigCont σ)) ∧
(z ∈ stmts_idents stmts) ∧
(evalVar y σ = evalVar z σ))))"

3.7.2 Control

Control assertion represents the object making a function call on another object. We give

a formalized definition of Calls, which goes along with supporting functions:

idents_list_undef and idents_list_equal.

Since the control assertion has a form x calls y.m(params), and the identifier x, y, as

well as all elements in the list params should be defined, we give an auxiliary function

idents_list_undef to check for any undefined identifiers in the list of identifiers, given

the configuration σ.

fun

idents_list_undef:: "Identifier list ⇒ Config ⇒ bool"

where

"idents_list_undef [] σ = False" |

"idents_list_undef (x#xs) σ =

((evalVar x σ = None) ∨ (idents_list_undef xs σ))"

Then, the function idents_list_equal checks that the value of each element of the first

and second identifier list are equal in runtime configuration σ.

fun

idents_list_equal:: "Identifier list ⇒ Identifier list ⇒ Config ⇒ bool"

where

"idents_list_equal [] [] σ = True" |

"idents_list_equal (z#zs) [] σ = False" |

"idents_list_equal [] (v#vs) σ = False" |

"idents_list_equal (z#zs) (v#vs) σ =

((evalVar z σ = evalVar v σ) ∧ (idents_list_equal zs vs σ))"

The assertion Calls x y m zs holds if in runtime configuration σ,

(1) the identifier x, y and all identifiers in list zs are defined, and

(2) the address of this object equals to the value of the caller x

(3) there are a receiver u and arguments vs with the same method m in runtime config-

uration σ such that the value of the identifier y and u is the same, and the value of each

element of the list zs and vs is equal.

definition

Calls ::

"Identifier ⇒ Identifier ⇒ MethodName ⇒ Identifier list ⇒ Assertion"

where

"Calls x y m zs ≡ λM M’ σ.

if (evalVar x σ = None ∨ evalVar y σ = None ∨
(idents_list_undef zs σ))

then None

else

Some (( thisEval σ = evalVar x σ) ∧
(∃ a u vs stmts.(ConfigCont σ =

Code (Seq (MethodCall a u m vs) stmts)) ∧
(evalVar y σ = evalVar u σ) ∧ idents_list_equal zs vs σ))"

3.7.3 Viewpoint

Viewpoint assertion represents whether an object belongs to the internal or external

module. The formalization of Internal and External can be found as follows.

First, we define a function Ident_class_lookup to look up the class where identifier x is

located in runtime configuration σ. The function returns None in case of failure.

fun

Ident_class_lookup :: "Config ⇒ Identifier ⇒ ClassName option"

where

"Ident_class_lookup σ x =

(case σ of (ϕ#ψ,χ) ⇒
(case (evalVar x σ) of

Some (VAddr a) ⇒ (case χ a of

Some obj ⇒ Some (className obj)

| None ⇒ None ))

| _ ⇒ None)"

Then, the assertion External x holds if the object x is outside the scope of module M in

configuration σ.

definition

External :: "Identifier ⇒ Assertion"

where

"External x ≡ λ M M’ σ.

(case (Ident_class_lookup σ x) of

None ⇒ None |

Some c ⇒ Some (c ∉ dom M))"

Otherwise, the assertion External x asserts that the object x is in module M in runtime

configuration σ.

definition

Internal :: "Identifier ⇒ Assertion"

where

"Internal x ≡ λ M M’ σ.

(case (Ident_class_lookup σ x) of

None ⇒ None |

Some c ⇒ Some (c ∈ dom M))"

3.7.4 Space

To define space assertion, we give a function restrictConf to create a new heap with only

objects in the given set S in runtime configuration σ.

definition

hRst :: "Heap ⇒ Config ⇒ Identifier ⇒ Heap"

where

"hRst χ σ S ≡
λa. (case (evalVar S σ) of

None ⇒ None |

Some v ⇒ (case v of VAddr addr ⇒ None |

VAddrSet addrSet ⇒ if a ∈ addrSet

then χ a

else None ))"

The function restrictConf updates the new heap χ’ after restricting the given set S in

runtime configuration σ.

definition

restrictConf :: "Identifier ⇒ Config ⇒ Config option"

where

"restrictConf S σ ≡
(case σ of (ψ, χ) ⇒ (let χ’ = (hRst χ σ S) in

Some (ψ, χ’)))"

definition

transConf :: "(Config ⇒ Config option) ⇒ Assertion ⇒ Assertion"

where

"transConf transf A ≡
λM M’ σ. (case transf σ of None ⇒ None | Some b ⇒ A M M’ b)"

Thanks to the restriction operator restrictConf, we obtain the semantics of the space

assertion In.

definition

In :: "Identifier ⇒ Assertion ⇒ Assertion"

where

"In S ≡ transConf (restrictConf S)"

After having the definition of space assertion In, we also provide lemmas related to spatial

connective assertions. Lemma Distrib_In proves the distributive property of space

assertion over logical implication.

lemma Distrib_In:

"(aImp (In S A) (In S B) M M’ σ) =

(In S(aImp A B)) M M’ σ"

by (simp add: In_def aImp_def bopt_def option.case_eq_if transConf_def)

Also, the lemma not_In demonstrates the negation property of space assertion.

lemma not_In:

"aNot (In S A) M M’ σ = (In S (aNot A)) M M’ σ"

proof -

have "∀f fa p fb fc. aNot (transConf fc fb) fa f p =

transConf fc (aNot fb) fa f p ∨ fc p = None"

using aNot_def transConf_def

by force

thus ?thesis

using In_def aNot_def transConf_def

by fastforce

qed

3.7.5 Adaptation on runtime configurations

This section is the most challenging part of giving a formalization. Thus, to define whether

a runtime configuration satisfies a time assertion, we need to adapt a runtime configura-

tion to another to deal with time.

To cope with the time concept, we encounter some challenges: (a) the validity of assertions

in the future must be evaluated in the future configuration but utilizing the current

configuration’s bindings. For example, the assertion Will(x.f = 1) is satisfied if the field

f of the object pointed at by x in the current configuration has the value 1 in some future

configuration. Note that x may be pointing to a different object in the future configuration

or may no longer be in scope. Therefore, the operator ◁ is used to combine runtime

configurations. In particular, σ ◁ σ’ adapts the following configuration to the view of top

frame view of the former, returning a new one whose stack has the top frame as received

from σ and where the cont has been renamed consistently, while the heap is taken from

σ’. It permits to interpret expressions in the newer configuration σ’ but with the variables

tied in keeping with the top frame from σ.

The second obstacle we need to grab is that (b) the current configuration requires to store

the code executed to determine future configurations. We cope with it by storing the

residual code in the continuation in each frame.

Next, (c) we do not desire to observe configurations beyond the frame at the top of the

stack. We handle it by only getting the top of the frame as pondering future executions.

We give a formalized definition of adaptation on runtime configuration as adaptation.

In the definition, we need support from adapt_frame σ σ’, returning a new frame that

consists of

(1) A new continuation: it is the same as the continuation of the configurationσ’. However,

we replace all variables zs with fresh names zs’ using cont_subst_list (cont ϕ’) zs

zs’. The set zs’ comes from fresh_idents (dom (vars ϕ)) zs, which is a function

generating a list of fresh identifiers where none of the new identifiers appear in dom (vars

ϕ) or zs.

(2) A combination of the variable map from the configuration σ with the variable map

from the configuration σ’ through the renaming vars ϕ(zs’ [ 7→] map (λz. the (vars

ϕ’ z)) zs).

We present all additional definitions and lemmas to formalize adaptation on runtime

configurations in Appendix A.4.

The function cont_subst_list replaces all variables zs with fresh names zs’.

fun

cont_subst_list ::

"Continuation ⇒ Identifier list ⇒ Identifier list ⇒ Continuation"

where

"cont_subst_list (Code stmts) zs zs’ =

(Code (stmts_subst_list stmts zs zs’))" |

"cont_subst_list (NestedCall x stmts) zs zs’ =

(NestedCall (ident_subst_list x zs zs’) (stmts_subst_list stmts zs zs’))"

definition

adapt_frame :: "Frame ⇒ Frame ⇒ Frame"

where

"adapt_frame ϕ ϕ’ ≡
(let zs = sorted_list_of_set (dom (vars ϕ’));

zs’ = fresh_idents (dom (vars ϕ)) zs;

contn’’ = cont_subst_list (cont ϕ’) zs zs’ ;

vars’’ = map_upds (vars ϕ) zs’ (map (λz. the ((vars ϕ’) z)) zs) in

(|cont = contn’’, vars = vars’’, this = (this ϕ’) |))"

The operator ◁ denotes adaptation between two runtime configurations, defined in func-

tion adaptation below.

definition

adaptation :: "Config ⇒ Config ⇒ Config option" (" _ ◁ _")

where

"σ ◁ σ’ ≡ (case σ of (ϕ#_,_) ⇒
(case σ’ of (ϕ’#ψ’,χ’) ⇒

let ϕ’’ = adapt_frame ϕ ϕ’ in

Some (ϕ’’#ψ’,χ’) | _ ⇒ None)

| _ ⇒ None)"

3.7.6 Time

With full support from the definition of Adaptation adaptation, we define Next assertion

Next and Will assertion Will.

The function next_visible, a partial function, is used to reach the next state of visible-

state semantics if it exists.

definition

next_visible :: "Module ⇒ Module ⇒ Config ⇒ Config option"

where

"next_visible M M’ σ ≡
if (∃σ’. (M;M’, σ →e σ’))

then Some (THE σ’. (M;M’, σ →e σ’))

else None"

Similarly, the function will_visible is used to reach the future state of visible-state

semantics when it existed.

definition

will_visible :: "Module ⇒ Module ⇒ Config ⇒ Config option"

where

"will_visible M M’ σ ≡
if (∃σ’. (M;M’, σ →e* σ’))

then Some (THE σ’. (M;M’, σ →e* σ’))

else None"

The assertion Next A holds if A holds in some configuration σ’ which arises from exe-

cution ϕ, where ϕ is the top frame of σ. By requiring that next_visible M M’ ([ϕ], χ)

rather than next_visible M M’ σ, we are restricting the set of possible next configura-

tions to those caused by the top frame.

definition

Next :: "Assertion ⇒ Assertion"

where

"Next A ≡
λM M’ σ. (case σ of (ϕ#_,χ) ⇒

(case (next_visible M M’ ([ϕ],χ)) of Some σ’ ⇒
(case (([ϕ],χ) ◁ σ’) of

Some adpt ⇒ (A M M’ adpt)|

None ⇒ None) |

None ⇒ None) |

_ ⇒ None)"

Similar to the assertion Next A, we define the assertion Will A. It says that the assertion

holds when A holds in some configuration σ’ which arises from execution ϕ, where ϕ is

the top frame of σ. However, it considers in more future steps instead of in the successive

step.

definition

Will :: "Assertion ⇒ Assertion"

where

"Will A ≡
λM M’ σ. (case σ of (ϕ#_,χ) ⇒

(case (will_visible M M’([ϕ],χ) ) of Some σ’ ⇒
(case (([ϕ],χ) ◁ σ’) of

Some adpt ⇒ (A M M’ adpt)|

None ⇒ None) |

None ⇒ None) |

_ ⇒ None)"

3.7.7 Authority

Authority (Changes) assertion Changes is defined to give conditions for change to occur.

The partial function Changes on expression e is used to assert the evaluation of expression

e in the next configuration σ’ distinguishes from one in the current configuration σ.

In particular, the Changes e says that there exists an expression v such that the value of

the expression v and e is the same. However, the expression v and e ’s values are no longer

the same in the next configuration.

definition

Changes :: "Expr ⇒ Assertion"

where

"Changes e ≡ λM M’ σ.

(case (next_visible M M’ σ) of Some σ’ ⇒
(case (σ ◁ σ’) of Some adpt ⇒

(case expr_eval e σ of Some v1 ⇒
(case expr_eval e adpt of Some v2 ⇒

Some (v1 ̸= v2) |

None ⇒ None ) |

None ⇒ None) |

None ⇒ None) |

None ⇒ None)"

3.7.8 Modules Satisfying Assertions

The section exhibits how we formally define whether a module satisfies an assertion

Module_sat. Here, Module_sat M A holds when for all external modules M’ and all Arising

runtime configuration σ, the assertion A is satisfied. Note that all runtime configuration σ

is observed as Arising configurations.

definition

Module_sat :: "Module ⇒ Assertion ⇒ bool"

where

"Module_sat M A ≡ (∀ M’ σ. (σ ∈ Arising M M’) −→
(A M M’ σ = None ∨ A M M’ σ = Some True))"

3.8 Summary

So far, we have formalized the theory of holistic specifications and several proofs com-

prising about 1800 lines of code of definitions and proofs in Isabelle/HOL. We have also

proved several lemmas related to the theory of holistic specifications, as follows.

1. Execution in Loo and visible states is deterministic. The deterministic execution of

the language Loo is shown at Lemma exec_det in Section 3.4.1. Also, the deter-

ministic execution of the visible states is proved at Lemma visible_exec_det in

Section 3.4.1.

2. Properties of Linking. The linking is associative, and commutative can be found

at Lemma link_assoc and Lemma link_commute in Section 3.3. Moreover, the

linking preserves both one-module and two-module execution is shown at Lemma

visible_exec_linking_1 and Lemma visible_exec_linking_2 in Section 3.4.2.

3. Assertions are classical. Lemmas of assertions related to standard logical operators

and quantifiers (∧, ∨, →, ¬, ∀ and ∃) are shown in Section 3.6.2. They include

aComplement_1, aComplement_2, aCommutative_1, aCommutative_2,

aAssociative_1, aAssociative_2, aDistributive_1, aDistributive_2,

aDeMorgan_1, aDeMorgan_2, aUniversal_existential_1,

aUniversal_existential_2, aImplication, and aNeverHold.





Chapter 4

Lemmas towards reasoning about

Holistic specifications

This chapter paves a way to reason about Holistic specifications. First, we briefly consider

an example to reflect that the traditional specification is insufficient to guarantee the code

is secure. Next, we construct a new specification as a form of holistic specifications to

make the code safer. Then, from the holistic specifications made earlier, we form lemmas

and provide proofs and “pen-and-paper” proofs to establish the foundations for reasoning

about holistic specifications.

Chapter Outline

• Motivating example. Section 4.1 provides the example of class Wallet and the

construction of its specifications as forms of holistic specifications.

• Lemmas for reasoning about holistic specifications. Section 4.2 presents lemmas

and proofs with hoping to place the foundations for reasoning about holistic specifi-

cations.



52 Lemmas towards reasoning about Holistic specifications

4.1 Motivating example

Let consider the code snippet from Figure 4.1. Class Wallet consists of a balance field, a

secret field, and the method pay, which takes a role as the only holder of the secret, can

use the Wallet to make payments – for the sake of simplicity, we allow balances to grow

negative.

1 class Wallet {

2 fld balance

3 fld secret

4 mthd pay(who, amt, scr) {

5 if (secret == src) {

6 balance = balance - amt

7 who.balance = who.balance + amt

8 }

9 }

10 }

Listing 4.1 Example of class Wallet modified from class Safe [8].

We give a Hoare triple specification below detailing the behavior of the method pay.

method pay(who, amt, scr)

PRE: (this, who:Wallet) ∧ (this ̸= who) ∧ (amt:N) ∧ (scr = secret)

POST: (this.balance= this.balancepre−amt) ∧
(who.balance= who.balancepre+amt)

(4.1)

The specification in Formula 4.1 shows that the secret is a sufficient condition to make a

payment. Namely, if the secret succeeds in providing, then the pay method can perform

a payment. Though, it does not show that the specification is a necessary condition. In

the case the pre-condition is not satisfied, how does the traditional Hoare triple represent

the behavior of the method pay? We also describe the behavior of the method pay if the

pre-condition is not satisfied, as follows.

method pay(who, amt, scr)

PRE: (this, who:Wallet) ∧¬((this ̸= who) ∧ (amt:N) ∧ (scr = secret))

POST: ∀w:Wallet.(w.balance= w.balancepr e −amt)
(4.2)

The specification in Formula 4.2 meets that if the secret fails to provide, then the method

pay cannot make a payment. However, the specification cannot prevent some other

classes (including Wallet) containing more methods, making it possible to affect a re-

duction in the balance without knowing the secret. To circumvent this, we form a new

specification using the concepts of holistic specification presented below in Formula 4.3.

Spec≜∀w,m.( w:Wallet∧w.balance = m∧Will(w.balance ̸= m) =⇒
∃o. (External(o)∧(o Access w.secret)))

(4.3)

The specification expresses that for any wallet w defined in the current configuration, if

the balance of w were to change in the future, then no less than one external object in the

current configuration has direct access to the secret.

Straightforwardly, the question here is when a module meets the holistic specification

outlined in Formula 4.3. In particular, the module is a consideration module (a.k.a.,

internal module), which consists of the code of the class Wallet from Figure 4.1.

Drossopoulou et al. [8] in Section 4.1 answered that M |= Spec holds if for all untrusted

modules M′ and all Arising configurationsσ of execution of code from module-pair (M,M′),

the Spec holds.

We have specified it formally as follows.

M |= Spec if ∀ M′. ∀σ ∈ Arising(M;M′).[M;M′,σ |= Spec] (4.4)

Based on definitions of holistic assertions defined in [8], we open Formula 4.4 with the

Spec with purposes to go deeply and from that to produce lemmas or theorems to reason-

ing the Spec.

First, we expand M;M′,σ |= Spec. It is equivalent to the formula below.

M;M′,σ |= [∀w,m.( w:Wallet∧w.balance = m∧Will(w.balance ̸= m) =⇒
∃o. (External(o)∧(o Access w.secret)))]

(4.5)

We have the below formula by employing assertions directly with logical connectives and

quantifiers.

∀w,m. (M;M′,σ |= w:Wallet ∧M;M′,σ |= (w.balance = m)∧
M;M′,σ |= Will(w.balance ̸= m)

=⇒ ∃o.
(
M;M′,σ |= External(o)∧M;M′,σ |= (o Access w.secret)

) (4.6)

We produce Formula 4.7 using the definition of Will assertion. Note that the operator ◁

denotes adaptation between two runtime configurations, formalized in Section 3.7.5.

M;M′,σ |= Will(w.balance ̸= m)
if ∃σ′.(M;M′,σ→∗

e σ
′∧M;M′,σ◁σ′ |= (w.balance ̸= m))

(4.7)

After having Formula 4.7, we produce Formula 4.8 from Formula 4.6 below.

∀w,m. (M;M′,σ |= w:Wallet ∧
M;M′,σ |= (w.balance = m)∧ (∃σ′.(M;M′,σ→∗

e σ
′))∧

∀o.
(
M;M′,σ |= (o Access w.secret) =⇒ M;M′,σ |= Internal(o)

)
=⇒ M;M′,σ◁σ′ |= (w.balance= m)

(4.8)

4.2 Lemmas for reasoning about holistic specifications

We understand that reasoning on temporal logic Will〈A〉 assertion is difficult to reason

about the future in more number steps time. Hence, we choose to reason in one step

instead of in more number of steps.

To show that reasoning in one step is still sufficient, we have a technical Theorem 1.

Assume that we have several finite steps, and instead of writing σ0 →∗
e σn , we also assign a

counter variable n. Let denote σ0 →n
e σn as a path σ0 →e σ1 →e σ2 →e · · ·→e σk · · ·→e σn ,

where 1 ≤ k ≤ n.

Theorem 1. Let σ0, σ, and σ′ be an initial configuration, and arbitrary configurations

respectively. Let S, A, and B be as follows.

S := P (σ0)∧ (σ0 →n
e σn)∧¬Q(σ0) =⇒ ¬W (σn),

A := P (σ)∧ (σ→e σ
′)∧¬Q(σ) =⇒ P (σ′)∧¬Q(σ′),

and

B := P (σ)∧ (σ→e σ
′)∧¬Q(σ) =⇒ ¬W (σ′).

We have A∧B ⊢ S.

Proof. We show the lemma using induction on n.

• Case n = 1.

We need to prove S1 := P (σ0)∧ (σ0 →1
e σ1)∧¬Q(σ0) =⇒ ¬W (σ1).

Since σ0 →e σ1, we have

A1 := P (σ0)∧ (σ0 →e σ1)∧¬Q(σ0) =⇒ P (σ1)∧¬Q(σ1), and

B1 := P (σ0)∧ (σ0 →e σ1)∧¬Q(σ0) =⇒ ¬W (σ1).

S1 is shown obviously from B1.

• Case n = 2.

We need to prove that S2 := P (σ0)∧ (σ0 →2
e σ2)∧¬Q(σ0) =⇒ ¬W (σ2).

From the left-hand side (LHS) of S2, we have σ0 →2
e σ2,

meaning σ0 →e σ1 →e σ2.

We have the transition from σ0 →e σ1; therefore, we have A1 and B1.

Now, let consider a transition from σ1 →e σ2.

Combining σ1 →e σ2 and A, B , and the right-hand side (RHS) of A1, we have

P (σ1)∧ (σ1 →e σ2)∧¬Q(σ1) =⇒ P (σ2)∧¬Q(σ2), and

P (σ1)∧ (σ1 →e σ2)∧¬Q(σ1) =⇒ ¬W (σ2).

From that, we can conclude

A2 := P (σ0)∧ (σ0 →2
e σ2)∧¬Q(σ0) =⇒ P (σ2)∧¬Q(σ2),

and B2 := P (σ0)∧ (σ0 →2
e σ2)∧¬Q(σ0) =⇒ ¬W (σ2).

Similarly, S2 is shown from B2.

• Assume the case n = k is true, we have

Sk := P (σ0)∧ (σ0 →k
e σk )∧¬Q(σ0) =⇒ ¬W (σk ).

Since the formula is true with the path from σ0 →e σ1 →e σ2 →e · · ·→e σk ,

we also have Ak := P (σ0)∧ (σ0 →k
e σk )∧¬Q(σ0) =⇒ P (σk )∧¬Q(σk ), and

Bk := P (σ0)∧ (σ0 →k
e σk )∧¬Q(σ0) =⇒ ¬W (σk ).

We need to prove with the case n = k +1.

• Case n = k +1.

We need to prove that Sk+1 := P (σ0)∧ (σ0 →k+1
e σk+1)∧¬Q(σ0) =⇒ ¬W (σk+1).

There is a transition from σk →e σk+1.

Combining σk →e σk+1 and A, B , and RHS of Ak , we have

P (σk )∧ (σk →e σk+1)∧¬Q(σk ) =⇒ P (σk+1)∧¬Q(σk+1), and

P (σk )∧ (σk →e σk+1)∧¬Q(σk ) =⇒ ¬W (σk+1).

From that, we can conclude

Ak+1 := P (σ0)∧ (σ0 →k+1
e σk+1)∧¬Q(σ0) =⇒ P (σk+1)∧¬Q(σk+1), and

Bk+1 := P (σ0)∧ (σ0 →k+1
e σk+1)∧¬Q(σ0) =⇒ ¬W (σk+1).

Similarly, Sk+1 is shown from Bk+1.

As a result, we have A∧B ⊢ S.

Thanks to Theorem 1, we only need to reason about the predicate A and B . Now, we

rewrite Formula 4.8 as the form of the predicate A in Formula 4.9 as follows.

∀w,m. (M;M′,σ |= w:Wallet ∧
M;M′,σ |= (w.balance = m)∧ (∃σ′.(M;M′,σ→e σ

′))∧
∀o.

(
M;M′,σ |= (o Access w.secret) =⇒ M;M′,σ |= Internal(o)

)
=⇒ M;M′,σ◁σ′ |= w:Wallet ∧M;M′,σ◁σ′ |= (w.balance = m) ∧

∀o.
(
M;M′,σ◁σ′ |= (o Access w.secret) =⇒ M;M′,σ◁σ′ |= Internal(o)

)
)

(4.9)

We also rewrite Formula 4.8 as the form of the predicate B in Formula 4.10 below.



4.2 Lemmas for reasoning about holistic specifications 57

∀w,m. (M;M′,σ |= w:Wallet ∧
M;M′,σ |= (w.balance = m)∧ (∃σ′.(M;M′,σ→e σ

′))∧
∀o.

(
M;M′,σ |= (o Access w.secret) =⇒ M;M′,σ |= Internal(o)

)
=⇒ M;M′,σ◁σ′ |= (w.balance= m)

(4.10)

Note that the runtime configuration σ′ in Formula 4.9 and 4.10 is different from the

runtime configuration σ′ in Formula 4.8.

From Formula 4.9 and 4.10, we set out lemmas with goals to support reasoning these

specifications. In a capability system, all access to a capability must derive from pre-

existing access. In short, “only connectivity begets connectivity” [22]. This thesis states

and proves the properties of “only connectivity begets connectivity” (Theorem 2 and

Theorem 3). We also form some technical lemmas aiding to prove them. While we have

finished proving Theorem 2, we have already proved a partial of Theorem 3.

Before going to the details of theorems, we want to remember high-level terms and some

interpretations throughout the upcoming sections. To be convenient, we frequently shift

among notations.

• Frames φ are mappings from identifiers xto values, where values include addresses.

• Heaps χ are mappings from addresses α to objects.

• Configurations σ are pairs (ψ,χ) where ψ is a list of stack frames φ and χ is the heap.

• Lookup of identifier x in runtime configuration σ is written ⌊x⌋σ. Lookup of identi-

fier x in the stack frame φ is written ⌊x⌋φ. Both ⌊x⌋σ and ⌊o⌋φ is defined as same as

φ(x).

• Lookup of field f from x object in runtime configuration σ is written ⌊x. f ⌋σ. Re-

sult of ⌊x. f ⌋σ is a value v and χ(φ(x)) = (C , f ld M ap), where f ld M ap( f ) = v with

f ld M ap is a mapping from field name to values.

• The operator◁ denotes adaptation between two runtime configurations, formalized

in Section 3.7.5.

• The term σ.cont is used to obtain the continuation cont from runtime configura-

tion σ.

Theorem 2 says that an object has a direct path to another object if and only if there is a

direct path on the same objects in the next configuration. Note that the lookup of objects

is the same as the lookup of other objects, in which the current configuration or the next

configuration is a part.

Theorem 2. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let o1,o2 be identifiers such that ⌊o1⌋σ = ⌊this⌋σ and ⌊o2⌋σ′ = ⌊this⌋σ′ .

Let o3, o4 be identifiers such that ⌊o3⌋σ, ⌊o4⌋σ ̸∈ {⌊o1⌋σ, ⌊o2⌋σ′}.

Also, let heaps be χ and χ′ such that χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show that σ |= Next (o3 access o4) ⇐⇒ σ |= (o3 Access o4).

Proof. To prove the lemma, we have two cases to consider as follows.

1. σ |= Next (o3 Access o4) =⇒ σ |= (o3 Access o4). We prove it in Lemma 3.

2. σ |= (o3 Access o4) =⇒ σ |= Next (o3 Access o4). We prove it in Lemma 4.

As a result, we have σ |= Next (o3 Access o4) ⇐⇒ σ |= (o3 Access o4).

Theorem 3 says that if an object has a direct path to another object in the next configura-

tion, then there is a direct path on the same objects in the current configuration. In this

case, the lookup of the first object is the same as the lookup of other objects. The current

configuration or the next configuration is a part, and the second object is in the current

configuration.

Theorem 3. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let oi , where i = 1,2 be identifers such that ⌊o1⌋σ = ⌊this⌋σ, ⌊o2⌋σ′ = ⌊this⌋σ′ , o2 ∈ dom(σ)

and ok ∈ dom(σ).

Also, let heaps be χ and χ′ such that χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show thatσ |= Next (oi access ok ) =⇒ (σ |= (o1 Access ok )) ∨ (σ |= (o2 Access ok )).

Note that we have already proved a partial of Theorem 3. Therefore, we put the proofs

in Appendix A.7.

Lemma 3 gives the “forwards” proof of Theorem 2, presented as follows.

Lemma 3. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let o1,o2 be identifiers such that ⌊o1⌋σ = ⌊this⌋σ and ⌊o2⌋σ′ = ⌊this⌋σ′ .

Let o3, o4 be identifiers such that ⌊o3⌋σ, ⌊o4⌋σ ̸∈ {⌊o1⌋σ, ⌊o2⌋σ′}.

Also, let heaps be χ and χ′ such that χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show that σ |= Next (o3 Access o4) =⇒ σ |= (o3 Access o4).

Proof. To be convenient, we call LHS of the implication is σ |= Next (o3 Access o4),

and RHS is σ |= (o3 Access o4).

From LHS of the implication of the formula, σ |= next (o3 Access o4) holds if

(σ◁σ′) |= (o3 Access o4) (1)

Using the definition of Permission in [8] (formalized in Section 3.7.2 as well), we rewrite

(1) as an equivalent form as follows.

(⌊o3⌋σ◁σ′ = ⌊o4⌋σ◁σ′)∨ (⌊o3. f ⌋σ◁σ′ = ⌊o4⌋σ◁σ′)∨

[(⌊o3⌋σ◁σ′ = ⌊this⌋σ◁σ′)∧ (⌊o4⌋σ◁σ′ = ⌊z′⌋σ◁σ′)], where z′ appears in (σ◁σ′).cont′.

• Case ⌊o3⌋σ◁σ′ = ⌊o4⌋σ◁σ′ .

We have ⌊o3⌋σ◁σ′ and ⌊o4⌋σ◁σ′ are defined in σ◁σ′.
Also, we have ⌊o3⌋σ◁σ′ = ⌊o3⌋σ, and ⌊o4⌋σ◁σ′ = ⌊o4⌋σ, since o3 and o4 are defined in

σ (See Lemma 7). Therefore, we have ⌊o3⌋σ = ⌊o4⌋σ in σ, and it implies o3 Access o4

in σ. From that, we have RHS of the implication of the formula.

• Case ⌊o3. f ⌋σ◁σ′ = ⌊o4⌋σ◁σ′ , with some field f .

The configuration σ◁σ′ and σ′ use the same heap χ′ (from Definition of Adaptation

in [8]). We rewrite ⌊o3. f ⌋σ◁σ′ = v for some v , and χ′(φ(o3)) = (C , f ld M ap), where

f ld M ap( f ) = v . On the other hand, we have ⌊o4⌋σ◁σ′ = ⌊o4⌋σ.

Now we need to show that ⌊o3. f ⌋σ = ⌊o3. f ⌋σ◁σ′ .

Thus, we have χ′(φ(o3)) =χ(φ(o3)), since Lemma 6 and ⌊o3⌋σ ̸= ⌊this⌋σ. Therefore,

⌊o3. f ⌋σ and ⌊o3. f ⌋σ◁σ′ are evaluated by the same value v .

So, we have ⌊o3. f ⌋σ = ⌊o4⌋σ, and it implies o3 Access o4 in σ.

Then, we also have RHS of the implication of the formula.

• Case (⌊o3⌋σ◁σ′ = ⌊this⌋σ◁σ′)∧(⌊o4⌋σ◁σ′ = ⌊z′⌋σ◁σ′), where z′ appears in (σ◁σ′).cont′.
Since σ◁σ′ has address φ′(this), we have ⌊this⌋σ◁σ′ = ⌊this⌋σ′ .

However, from ⌊o3⌋σ, ⌊o4⌋σ ̸∈ {⌊o1⌋σ, ⌊o2⌋σ′}, ⌊o1⌋σ = ⌊this⌋σ and ⌊o2⌋σ′ = ⌊this⌋σ′ .

There is a contradiction here.

As a result, we have σ |= Next (o3 Access o4) =⇒ σ |= (o3 Access o4).

Lemma 4 gives the “backwards” proof of Theorem 2, presented as follows.

Lemma 4. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let o1,o2 be identifiers such that ⌊o1⌋σ = ⌊this⌋σ and ⌊o2⌋σ′ = ⌊this⌋σ′ .

Let o3, o4 be identifiers such that ⌊o3⌋σ, ⌊o4⌋σ ̸∈ {⌊o1⌋σ, ⌊o2⌋σ′}.

Also, let heaps be χ and χ′ such that χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show that σ |= (o3 Access o4) =⇒ σ |= Next (o3 Access o4).

Proof. To be convenient, we call LHS is σ |= (o3 Access o4),

and RHS is σ |= Next (o3 Access o4) of the implication.

From LHS of the implication of the formula, σ |= (o3 Access o4) holds if

σ |= (o3 Access o4)

We rewrite the formula above as an equivalent form as follows.

(⌊o3⌋σ = ⌊o4⌋σ)∨ (⌊o3. f ⌋σ = ⌊o4⌋σ)∨ [(⌊o3⌋σ = ⌊this⌋σ◁σ′)∧ (⌊o4⌋σ = ⌊z′⌋σ)],

where z′ appears in σ.cont.

• Case ⌊o3⌋σ = ⌊o4⌋σ.

We have ⌊o3⌋σ◁σ′ = ⌊o3⌋σ, and ⌊o4⌋σ◁σ′ = ⌊o4⌋σ, since o3 and o4 are defined in σ (See

Lemma 7).

Therefore, we have ⌊o3⌋σ◁σ′ = ⌊o4⌋σ◁σ′ inσ◁σ′, and it implies o3 Access o4 inσ◁σ′.
From that, we have RHS of the implication of the formula.

• Case ⌊o3. f ⌋σ = ⌊o4⌋σ, with some field f .

We rewrite ⌊o3. f ⌋σ = v for some v , and χ(φ(o3)) = (C , f ld M ap),

where f ld M ap( f ) = v .

On the other hand, since o4 is defined in σ, we have ⌊o4⌋σ = ⌊o4⌋σ◁σ′ .

The configuration σ◁σ′ and σ′ use the same heap χ′.
Now we need to show that ⌊o3. f ⌋σ = ⌊o3. f ⌋σ◁σ′ . Thus, we have χ(φ(o3)) =χ′(φ(o3)),

since Lemma 6 and σ(o3) ̸=σ(this).

Therefore, ⌊o3. f ⌋σ and ⌊o3. f ⌋σ◁σ′ are evaluated by the same value v . So, we have

⌊o3. f ⌋σ◁σ′ = ⌊o4⌋σ◁σ′ , implying o3 Access o4 in σ◁σ′. Then, we also have RHS of

the implication of the formula.

• Case (⌊o3⌋σ = ⌊this⌋σ)∧ (⌊o4⌋σ = ⌊z′⌋σ), where z′ appears in σ.cont.

We have ⌊o3⌋σ ̸= ⌊this⌋σ because ⌊o3⌋σ ̸= ⌊o1⌋σ and ⌊o1⌋σ = ⌊this⌋σ. There is a

contradiction here.

As a result, we have σ |= (o3 Access o4) =⇒ σ |= Next (o3 Access o4).

Lemma 5, 6, 7, 8, and 9 are technical lemmas supporting to prove Theorem 2 and 3, showed

as follows.

Lemma 5. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let field_update((C , f ld M ap), f , v) = (C , f ld M ap( f := v)) be a function using to up-

date a field.

Show that if χ′(φ(this)) ̸=χ(φ(this)) and φ(this) ∈ dom(χ),

then there exists a field f and an identifier x where identifier x ∈ dom(σ) such that

χ′(φ(this)) = field_update(χ(φ(this), f ,φ(x))).

Proof. By structural induction on operational semantics, we have

• Case methCall_OS. The heap χ is unchanged in the next configuration σ′.
It means χ′(φ(this)) =χ(φ(this)). Hence, the desired result is correct.

• Case varAssgn_OS. It is similar to the case of methCall_OS.

• Case return_OS. It is also similar to the case of methCall_OS.

• Case fieldAssgn_OS. The heap χ is only changed in σ′ if there is an identifier x

and a field f such that this. f := x in σ.cont. In particular, if there is this. f := x in

σ.cont, the new heap χ′ now is the heap χ at the address φ(this), and the field f is

now updated by φ(x).

Formally, χ′(φ(this)) =χ[φ(this) 7→ (C , f ld M ap[ f 7→φ(x)])]. So the desired result

is correct as well.

• Case objCreate_OS. The heap χ is also changed in σ′. However, the heap is only

changed at a new address α, where χ(α) = (C , f1 7→ φ(x1), . . . fn 7→ φ(xn)). Since

a ̸∈ dom(χ), and φ(this) ∈ dom(χ), we have α ̸=φ(this). From that, we still have

got χ′(φ(this)) =χ(φ(this)). So, the desired result is correct.

As a result, the lemma is proved.

Lemma 6. Let M ,σ→e σ
′ be an execution. Also, let an address a =σ(this).

Show that if for all address a′ such that a′ ̸= a, a′ ∈ dom(χ), and φ(this) ∈ dom(χ),

then χ′(a′) =χ(a′).

Proof. By structural induction on operational semantics,

• Case methCall_OS. The heap χ is unchanged in the next configuration σ′. It means

χ′(a′) =χ(a′). Hence, the desired result is correct.

• Case varAssgn_OS. It is similar to the case of methCall_OS.

• Case return_OS: It is also similar to the case of methCall_OS.

• Case fieldAssgn_OS. The heap χ only is updated if the field at the address φ(this)

changed. However, we consider the address a′ ∈ dom(χ) and a′ ̸=σ(this). There-

fore, there is no update in the heap here, and χ′(a′) =χ(a′). So, the desired result is

correct, as well.

• Case objCreate_OS. The heap χ is also changed in σ′. However, the heap is only

changed at a new address α, where χ(α) = (C , f1 7→φ(x1), . . . fn 7→φ(xn)). Since α is

new in the heap χ, we haveα ̸∈ dom(χ). So, a′ ̸=α. From that, we have χ′(a′) =χ(a′).

So, the desired result is also correct.

As a result, we have χ′(a′) =χ(a′).

Lemma 7. Let w be an identifier such that ⌊w⌋σ is defined. Show that ⌊w⌋σ◁σ′ = ⌊w⌋σ.

Proof. From Definition 8 of Adaptation on runtime configurations in [8], we haveβ2(zs1) =
β1(zs1), where β1 and β2 are variable maps from the runtime configuration σ1 and σ2

respectively, and zs1 is a set of identifiers defined in the current configuration σ.

Since w is defined in the current configuration σ, so we have ⌊w⌋σ◁σ′ = ⌊w⌋σ.

Lemma 8. Let w be an identifier such that w is fresh in σ and σ′.
Show that if ∃v. ⌊v⌋σ′ is defined, then ⌊w⌋σ◁σ′ = ⌊v⌋σ′ .

Proof. From Definition 8 of Adaptation on runtime configurations in [8], we haveβ2(zs′) =
β2(zs2), where β1 and β2 are variable maps from the runtime configuration σ and σ′

respectively, zs′ is a set whose all identifiers are fresh in both variable maps β1 and β1,

and zs2 is a set of identifiers defined in the current configuration σ′.

Since w is fresh in both the current configuration σ and the next configuration σ′, so there

exists an identifier v in zs2 such that ⌊w⌋σ◁σ′ = ⌊v⌋σ′ .

Lemma 9. For any identifier w ∈ dom(β′′), where β′′ is a variable map from σ◁σ′.
Show that either

(1) If w ∈ dom(β), then ⌊w⌋σ◁σ′ = ⌊w⌋σ, or

(2) If w ̸∈ dom(β) and w ̸∈ dom(β′), then ∃v.v ∈ dom(β′) such that ⌊w⌋σ◁σ′ = ⌊v⌋σ′ .

Proof. We obtain the proof with the use of Lemma 7 and Lemma 8.

All technical lemmas mentioned at 5, 6, 7, 8, and 9 are shown in Isabelle/HOL in Section

A.6 of the previous chapter.





Chapter 5

Conclusion and Future Work

5.1 Future Work

So far, we have a framework for holistic specifications formalized in Isabelle/HOL. The

formalization can be a foundation to extend it or settle other parts built on top of it. In

particular, we have some straightforward possible plans as follows.

• In this thesis, we have provided an Isabelle/HOL mechanization of the core of Chain-

mail. However, our formalization currently supports the only formal foundation

of holistic specifications. Therefore, the immediate plan could provide a formal

specification and verification in Isabelle/HOL some small case studies, i.e., Wallet
example mentioned at the beginning of Chapter 4. Next, we want to provide holistic

specifications and the reasoning of the sequence examples taken from the object-

capability literature, such as the Bank/Account example [22] specified in Section

2.1.2 or attenuating the DOM (Domain Object Model). Then, we develop a full logic

to prove their soundness in Isabelle/HOL.

• In a long-term plan, we plan to develop a technique of automated reasoning with

that logic suggested above. Then, inspired by the initial Bank/Account work, we

study the verification of capabilities in other programming languages, such as

Javascript or Solidity.

5.2 Conclusion

In this thesis, we have presented the problem of holistic specifications, the background

of Isabelle/HOL, and holistic specification, and a thorough survey of related work on the

object-capability model and verification for object-capability programs in Chapter 2. The

core of holistic specification language Chainmail is formalized in Isabelle/HOL mecha-

nization written down in Chapter 3. We have constructed an Isabelle/HOL mechanization

for the underlying language with proving the deterministic execution of this language,

the syntax and semantics for Chainmail, and the proofs of properties of the underlying

language, as well as Chainmail properties. We also have proposed lemmas and provided

informal or formal proofs related to the properties of “only connectivity begets connec-

tivity”, one of the two access principles enunciated in the object-capability literature in

Chapter 4. It is considered preliminary results towards reasoning about the capability

policies to be considered a fundamental means of verifying holistic specifications.



Appendix A

Auxiliary Functions, Lemmas in

Isabelle/HOL, and Partial Proofs of

Theorem 3

A.1 Auxiliary Functions supporting Operational semantics
theory Support

imports Main

begin

lemma Greatest_eq [simp]:

"(GREATEST x. x = (a::nat)) = a"

by (simp add: Greatest_equality)

lemma sorted_head:

"sorted xs =⇒ x ∈ set xs =⇒ hd xs ≤ x"

by(induction xs arbitrary: x, auto)

lemma sorted_last:

"sorted xs =⇒ x ∈ set xs =⇒ last xs ≥ x"

by(induction xs arbitrary: x, auto)

lemma last_sorted_list_of_list_is_greatest:

assumes fin: "finite A"

assumes yin: "(y::nat) ∈ A"

shows "y ≤ last (sorted_list_of_set A)"

by (simp add: fin sorted_last yin)

lemma not_eq_a:

"finite A =⇒ Suc (last (sorted_list_of_set (insert a A))) ̸= a"

by (metis Suc_n_not_le_n finite.insertI insertI1

last_sorted_list_of_list_is_greatest)

lemma not_in_A:

"finite A =⇒ Suc (last (sorted_list_of_set (insert a A))) ∉ A"

by (meson Suc_n_not_le_n finite.simps insertI2

last_sorted_list_of_list_is_greatest)

end

A.2 Technical Lemmas supporting Deterministic
theory Deterministic_Aux

imports Language

begin

lemma exec_det_aux:

"M, σ →e σ’ =⇒ M, σ →e σ’’ =⇒ σ’’ = σ’"

proof (induction arbitrary: σ’’ rule: exec.induct)

case (exec_method_call ϕ x y m params stmts a χ C paramValues M meth ϕ’’ ψ)

from 〈M, (ϕ # ψ, χ) →e σ’’ 〉

show ?case

apply(rule exec.cases)

using exec_method_call

by auto

next

case (exec_var_assign ϕ x y stmts M ψ χ)

from 〈M, (ϕ # ψ, χ) →e σ’’ 〉

show ?case

apply (rule exec.cases)

by (auto simp: exec_var_assign)

next

case (exec_field_assign ϕ y x stmts v χ χ’ M ψ)

from 〈M, (ϕ # ψ, χ) →e σ’’ 〉

show ?case

apply (rule exec.cases)

using exec_field_assign

by auto

next

case (exec_new ϕ x C params stmts paramValues M c obj’ a χ χ’ ψ)

from 〈M, (ϕ # ψ, χ) →e σ’’ 〉

show ?case

apply (rule exec.cases)

using exec_new

by auto

next

case (exec_return ϕ x stmts ϕ’ x’ stmts’ M ψ χ)

from 〈M, (ϕ # ϕ’ # ψ, χ) →e σ’’ 〉

show ?case

apply (rule exec.cases)

using exec_return

by auto

qed

lemma internal_exec_rev_tr_nonempty’:

"internal_exec_rev M M’ σ tr σ’ =⇒ tr = [] =⇒ False"

apply(induction rule: internal_exec_rev.induct, auto)

done

lemma internal_exec_rev_tr_nonempty:

"internal_exec_rev M M’ σ [] σ’ =⇒ False"

using internal_exec_rev_tr_nonempty’

by blast

lemma internal_exec_rev’_appI:

"internal_exec_rev’ M M’ σ tr σ’ =⇒
internal_exec_rev’ M M’ σ’ tr’ σ’’ =⇒
internal_exec_rev’ M M’ σ (tr @ tr’) σ’’"

apply(induct rule: internal_exec_rev’.induct)

using internal_step

by auto

lemma internal_exec_rev_appI:

assumes "internal_exec_rev M M’ σ tr σ’’’"

assumes "internal_exec_rev’ M M’ σ’’’ tr’ σ’’"

shows "internal_exec_rev M M’ σ (tr @ tr’) σ’’"

using assms

apply(induction tr arbitrary: σ σ’’’ tr’ σ’’)

using internal_exec_rev_tr_nonempty

apply blast

apply(erule internal_exec_rev.cases)

by (simp add: internal_exec_rev’_appI internal_exec_rev_first_step)

lemma trace_rev1:

"internal_exec M M’ σ tr σ’ =⇒ internal_exec_rev M M’ σ tr σ’"

proof (induction rule: internal_exec.induct)

case (internal_exec_first_step σ σ’ c c’)

thus ?case

by (simp add: internal_exec_rev_first_step internal_refl)

next

case (internal_exec_more_steps σ tr σ’ σ’’ c)

note facts = internal_exec_more_steps

thus ?case

proof (cases rule: internal_exec.cases)

case (internal_exec_first_step c c’)

thus ?thesis

using facts internal_exec_rev_first_step internal_refl internal_step

by auto

next

case (internal_exec_more_steps tr’ σ’’’ c’)

thus ?thesis

using internal_exec_rev_appI facts internal_refl internal_step

by blast

qed

qed

lemma internal_exec_tr_nonempty:

"internal_exec M M’ σ tr σ’ =⇒ tr = [] =⇒ False"

by(induct rule: internal_exec.induct, auto)

lemma internal_exec_tr_nonempty’ [simp]:

"internal_exec M M’ σ [] σ’ = False"

using internal_exec_tr_nonempty

by blast

lemma internal_exec_appI:

assumes "internal_exec_rev’ M M’ σ’ tr σ’’"

"M;M’, σ →ei 〈tr’ 〉 σ’"

shows "M;M’,σ →ei 〈tr’ @ tr 〉 σ’’"

using assms

proof (induction arbitrary: tr’ σ rule: internal_exec_rev’.induct )

case (internal_refl σ)

thus ?case by simp

next

case (internal_step σ σ’ c c’ tr σ’’)

thus ?case

proof -

have "M;M’,σ →ei 〈(tr’ @ [σ’]) @ tr 〉 σ’’"

by (meson internal_exec.simps internal_step internal_step)

thus ?thesis

by simp

qed

qed

lemma internal_exec_appI1:

assumes "internal_exec M M’ σ tr σ’ "

"internal_exec_rev’ M M’ σ’ tr’ σ’’ "

shows "internal_exec_rev M M’ σ (tr @ tr’) σ’’"

using assms internal_exec_rev_appI trace_rev1

by blast

lemma trace_rev2:

"internal_exec_rev M M’ σ tr σ’ =⇒ internal_exec M M’ σ tr σ’ "

proof (induction rule: internal_exec_rev.induct)

case (internal_exec_rev_first_step σ σ’ c c’ tr σ’’)

note facts = internal_exec_rev_first_step

from facts(7)

show ?case

proof(cases rule: internal_exec_rev’.cases)

case internal_refl

thus ?thesis

using internal_exec_first_step facts

by blast

next

case (internal_step σ’’’ c c’ tr’)

thus ?thesis

using internal_exec_appI internal_exec_first_step

internal_exec_rev_first_step

by (metis (full_types) append_Cons self_append_conv2)

qed

qed

lemma trace_rev:

"internal_exec_rev M M’ σ tr σ’ = internal_exec M M’ σ tr σ’ "

using trace_rev1 trace_rev2

by blast

lemma internal_exec_rev’_det_prefix:

"internal_exec_rev’ M M’ σ tr σ’ =⇒
internal_exec_rev’ M M’ σ tr’ σ’’ =⇒
(∃tr’’. (tr = tr’ @ tr’’) ∨ (tr’ = tr @ tr’’))"

proof(induction arbitrary: tr’ σ’’ rule: internal_exec_rev’.induct)

case (internal_refl σ)

thus ?case by blast

next

case (internal_step σ σ’ c c’ tr σ’’)

note facts = internal_step

from internal_step(8)

show ?case

proof(cases rule: internal_exec_rev’.cases)

case internal_refl

thus ?thesis

by blast

next

case (internal_step σ c c’ tr)

thus ?thesis



A.2 Technical Lemmas supporting Deterministic 73

using exec_det_aux facts

by (metis append_Cons)

qed

qed

lemma internal_exec_rev_det_prefix’:

"M;M’, σ →eir 〈tr 〉 σ’ =⇒
M;M’, σ →eir 〈tr’ 〉 v =⇒
(∃ tr’’.(tr = (tr’ @ tr’’)) ∨ (tr’ = tr @ tr’’))"

proof(induction arbitrary: tr’ v rule: internal_exec_rev.induct)

case (internal_exec_rev_first_step σ σ’ c c’ tr σ’’)

note facts = internal_exec_rev_first_step

from 〈M;M’,σ →eir 〈tr’ 〉 v 〉

show ?case

proof (cases rule: internal_exec_rev.cases)

case (internal_exec_rev_first_step σ’’’ c’’ c’’’ tr’’)

have "σ’’’ = σ’"

using facts(2) internal_exec_rev_first_step(3) exec_det_aux

by blast

with facts internal_exec_rev_first_step

show ?thesis

using internal_exec_rev’_det_prefix

by auto

qed

qed

lemma internal_exec_det_prefix’:

"M;M’, σ →ei 〈tr 〉 σ’ =⇒
M;M’, σ →ei 〈tr’ 〉 v =⇒
(∃ tr’’.(tr = tr’ @ tr’’) ∨ (tr’ = tr @ tr’’))"

using internal_exec_rev_det_prefix’ trace_rev

by blast

inductive_cases internal_exec_elim [elim!]: "internal_exec M M’ σ tr σ’’"

inductive_cases visible_exec_elim [elim!]: "visible_exec M M’ σ σ’’"

lemma internal_exec_det_aux:

"M;M’, σ →ei 〈tr 〉 σ’ =⇒ M;M’, σ →ei 〈tr 〉 v =⇒ v = σ’"

by (metis internal_exec_elim last_ConsL last_snoc)

lemma internal_exec_is_internal:

"internal_exec M M’ σ tr σ’ =⇒
∀σi ∈ set tr.(∃ c.(this_class_lookup σi = Some c ∧ c ∈ dom M))"

by (induction rule: internal_exec.induct) simp+

lemma internal_exec_appD:

assumes "M;M’,σ →ei 〈tr’ 〉 σ’"

shows "∀tr’’ σi. (M;M’,σ →ei 〈tr’ @ tr’’ 〉 σi) −→
(internal_exec_rev’ M M’ σ’ tr’’ σi)"

using assms

proof (induction tr’ arbitrary: σ σ’ rule: rev_induct)

case Nil

thus ?case by simp

next

case (snoc x xs)

note facts = internal_exec_more_steps

fix tr’’ σi

have x_eq [simp]: "x = σ’"

using snoc.prems

by auto

hence "M;M’,σ →ei 〈xs @ [σ’] 〉 σ’"

using snoc.prems

by blast

from 〈 M;M’,σ →ei 〈xs @ [x] 〉 σ’ 〉

show ?case

proof (cases rule: internal_exec.cases)

case (internal_exec_first_step c c’)

hence xs_Nil [simp]: "xs = []"

by simp

show ?thesis

proof(intro allI impI)

fix tr’’ σi

assume "M;M’,σ →ei 〈(xs @ [x]) @ tr’’ 〉 σi"

hence "M;M’,σ →ei 〈x # tr’’ 〉 σi"

by simp

hence "internal_exec_rev M M’ σ (x#tr’’) σi"

using trace_rev

by simp

thus "M;M’,σ’ →eir1 〈tr’’ 〉 σi"

apply (rule internal_exec_rev.cases)

by simp

qed

next

case (internal_exec_more_steps tr σ’’ c)

note facts = internal_exec_more_steps

hence xs_tr [simp]: "xs = tr"

by blast

have internal_xs: "M;M’,σ →ei 〈tr @ [σ’] 〉 σ’"

using snoc.prems

by auto

from this

obtain σtr

where "(M;M’,σ →ei 〈tr 〉 σtr) ∧
(internal_exec_rev’ M M’ σtr [σ’] σ’)"

using snoc(1) xs_tr internal_exec_more_steps

by metis

with snoc(1) have

IH: "∀tr’’ σi. (M;M’,σ →ei 〈xs @ tr’’ 〉 σi)

−→ (M;M’,σtr →eir1 〈tr’’ 〉 σi)"

by (metis xs_tr)

show ?thesis

proof(intro allI impI)

fix tr’’ σi

assume "M;M’,σ →ei 〈(xs @ [x]) @ tr’’ 〉 σi"

hence "M;M’,σ →ei 〈xs @ (x # tr’’) 〉 σi"

by simp

with IH have "M;M’,σtr →eir1 〈x # tr’’ 〉 σi"

by metis

thus " M;M’,σ’ →eir1 〈tr’’ 〉 σi"

apply (cases rule: internal_exec_rev’.cases)

using x_eq

by blast

qed

qed

qed

lemma internal_exec_appD1:

assumes "M;M’,σ →ei 〈tr’ 〉 σ’"

"M;M’,σ →ei 〈tr’ @ tr’’ 〉 σi"

shows "internal_exec_rev’ M M’ σ’ tr’’ σi"

using assms internal_exec_appD

by metis

lemma visible_exec_det_aux:

"M;M’, σ →e σ’ =⇒ link_wf M M’ =⇒
M;M’, σ →e σ’’ =⇒ σ’’ = σ’"

proof(induction arbitrary: σ’’ rule: visible_exec.induct)

case (visible_exec_intro M M’ σ tr σi σ’ c)

note facts = visible_exec_intro

from 〈M;M’,σ →e σ’’ 〉

show ?case

proof(cases rule: visible_exec.cases)

case (visible_exec_intro tr’ σi’ c)

note facts1 = visible_exec_intro

from 〈M;M’,σ →ei 〈tr 〉 σi 〉 have

"∀σi ∈ set tr.

(∃ c. this_class_lookup σi = Some c ∧ c ∈ dom M)"

using internal_exec_is_internal

by auto

from 〈M;M’,σ →ei 〈tr’ 〉 σi’ 〉 have

"∀σi ∈ set tr’.

(∃ c. this_class_lookup σi = Some c ∧ c ∈ dom M)"

using internal_exec_is_internal

by auto

obtain tr’’ where tr: "tr = tr’ @ tr’’ ∨ tr’ = tr @ tr’’"

using facts(1) visible_exec_intro(1) internal_exec_det_prefix’

by metis

hence "tr’’ = []"

proof (rule disjE)

show " tr = tr’ @ tr’’ =⇒ tr’’ = []"

proof (rule ccontr)

assume tr_def: "tr = tr’ @ tr’’" "¬(tr’’ = [])"

from 〈M;M’,σ →ei 〈tr 〉 σi 〉

have "M;M’,σ →ei 〈tr’ @ tr’’ 〉 σi"

using tr_def

by simp

from this and 〈M;M’,σ →ei 〈tr’ 〉 σi’ 〉

have inter_ex: "internal_exec_rev’ M M’ σi’ tr’’ σi"

using internal_exec_appD1

by auto

hence "tr’’ = []"

proof (cases rule: internal_exec_rev’.cases)

case internal_refl

thus ?thesis

by simp

next

case (internal_step σ’ c’ c’’ tr)

from facts1(2) internal_step(2)

have a1: "σ’’ = σ’"

using exec_det_aux

by auto

from facts1 internal_step

have a2: "this_class_lookup σ’’ = Some c ∧ c ∈ dom M’" and

a3: "this_class_lookup σ’ = Some c’’ ∧ c’’ ∈ dom M"

apply blast

using local.internal_step(5) local.internal_step(6)

by auto

have a4: "c ∈ dom M’ ∧ c’’ ∈ dom M"

by (simp add: local.internal_step(6)

local.visible_exec_intro(4))

from a1 a2 a3

have a5: " c = c’’"

by simp

from a4 a5 show ?thesis

using link_wf_def visible_exec_intro.prems(1)

by auto

qed

thus False

using 〈tr’’ ̸= [] 〉

by auto

qed

show "tr’ = tr @ tr’’ =⇒ tr’’ = []"

proof (rule ccontr)

assume tr_def1: "tr’ = tr @ tr’’" "¬(tr’’ = [])"

from 〈M;M’,σ →ei 〈tr’ 〉 σi’ 〉

have "M;M’,σ →ei 〈tr @ tr’’ 〉 σi’"

using tr_def1 by simp

from this and 〈M;M’,σ →ei 〈tr 〉 σi 〉

have inter_ex1: "internal_exec_rev’ M M’ σi tr’’ σi’"

using internal_exec_appD1

by auto

hence "tr’’ = []"

proof (cases rule: internal_exec_rev’.cases)

case internal_refl

thus ?thesis

by simp

next

case (internal_step σ’ c’ c’’ tr)

thus ?thesis using facts exec_det_aux link_wf_def

by (metis IntI empty_iff option.inject)

qed

thus False

using 〈tr’’ ̸= [] 〉 by auto

qed

qed

thus ?thesis

using facts1 tr exec_det_aux append.right_neutral

internal_exec_det_aux visible_exec_intro.hyps(1)

visible_exec_intro.hyps(2)

by metis

qed

qed

end

A.3 Technical Lemmas supporting Linking module preserv-

ing execution
theory LinkingPreserve_Aux

imports Deterministic

begin

lemma link_exec_aux:

" [[ M, σ →e σ’; link_wf M M’ ]] =⇒ (M ◦l M’), σ →e σ’"

unfolding link_wf_def moduleLinking_def

moduleAux_def dom_def build_call_frame_def

proof (induction rule: exec.induct)

case (exec_method_call ϕ x y m params stmts α χ C paramValues M meth ϕ’’ ψ)

thus ?case

by (simp add: exec.exec_method_call ident_lookup.induct

M_def split: option.splits)

next

case (exec_var_assign ϕ x y stmts M ψ χ)

thus ?case

apply (simp add: this_field_lookup.induct)

using exec.exec_var_assign by auto

next

case (exec_field_assign ϕ y x stmts v χ χ’ M ψ)

thus ?case

apply (simp add: this_field_lookup.induct ident_lookup.induct)

by (simp add: exec.exec_field_assign)

next

case (exec_new ϕ x C params stmts paramValues M c obj’ α χ χ’ ψ)

thus ?case

by (simp add: ident_lookup.induct exec.exec_new)

next

case (exec_return ϕ x stmts ϕ’ x’ stmts’ M ψ χ)

thus ?case

apply (simp add: ident_lookup.induct)

using exec.exec_return by auto

qed

lemma internal_linking_1_aux:

" [[M;M’,σ →ei 〈tr 〉 σ’; link_wf_3M M M’ M’’ ]] =⇒
M; (M’ ◦l M’’), σ →ei 〈tr 〉 σ’"

proof (induction rule: internal_exec.induct)

case (internal_exec_first_step σ σ’ c c’)

have a: "link_wf M (M’ ◦l M’’)"

using 〈link_wf_3M M M’ M’’ 〉 link_wf_def link_wf_3M_def

by fastforce

have wf: "link_wf M M’"

using 〈link_wf_3M M M’ M’’ 〉

by simp

have "((M ◦l M’) ◦l M’’), σ →e σ’"

apply(rule link_exec_aux)

apply(rule 〈(M ◦l M’), σ →e σ’ 〉)

using 〈(link_wf_3M M M’ M’’) 〉 link_wf_3M_dest

by blast

from this wf link_assoc

have b: "(M ◦l (M’ ◦l M’’)), σ →e σ’"

by metis

have "dom (M’ ◦l M’’) = dom M’ ∪ dom M’’"

by (simp add: 〈link_wf_3M M M’ M’’ 〉)

hence c: "this_class_lookup σ = Some c ∧ c ∈ dom (M’ ◦l M’’)"

using internal_exec_first_step

by blast

from a b c internal_exec_first_step

show ?case

using internal_exec.internal_exec_first_step

by blast

next

case (internal_exec_more_steps σ tr σ’ σ’’ c)

have a: "M;M’ ◦l M’’,σ →ei 〈tr 〉 σ’ "

using 〈link_wf_3M M M’ M’’ =⇒ M;M’ ◦l M’’,σ →ei 〈tr 〉 σ’ 〉

and 〈link_wf_3M M M’ M’’ 〉

by simp

have asm: "M ◦l (M’ ◦l M’’) = (M ◦l M’) ◦l M’’"

by (metis internal_exec_more_steps.prems

link_assoc link_wf_3M_dest(1))

have "((M ◦l M’) ◦l M’’), σ’ →e σ’’ "

apply (rule link_exec_aux)

apply (rule 〈(M ◦l M’), σ’ →e σ’’ 〉)

using 〈link_wf_3M M M’ M’’ 〉 link_wf_3M_dest

by blast

from this and asm

have b: "(M ◦l (M’ ◦l M’’)), σ’ →e σ’’"

by simp

have c: "this_class_lookup σ’’ = Some c ∧ (c ∈ dom M)"

using internal_exec_more_steps

by simp

from a b c internal_exec.internal_exec_more_steps

show ?case

by simp

qed

lemma internal_linking_2_aux:

"M;M’,σ →ei 〈tr 〉 σ’ =⇒ link_wf_3M M M’ M’’=⇒
(M ◦l M’’); M’, σ →ei 〈tr 〉 σ’"

proof (induction rule: internal_exec.induct)

case (internal_exec_first_step σ σ’ c c’)

have a: "link_wf (M ◦l M’’) M’"

using 〈link_wf_3M M M’ M’’ 〉

by (fastforce simp add: link_wf_def link_wf_3M_def)

have " M◦l M’’ = M’’◦l M "

using 〈link_wf_3M M M’ M’’ 〉 link_commute

by blast

hence "(M◦l M’’) ◦l M’ = M’’◦l (M ◦l M’)"

using 〈link_wf_3M M M’ M’’ 〉 link_wf_3M_def link_wf_def

internal_exec_first_step.hyps(1)

internal_exec_first_step.prems

link_assoc link_commute link_wf_3M_dest(2)

link_wf_3M_dest(3) link_wf_3M_dest(4)

by metis

hence "(M◦l M’’) ◦l M’ = (M ◦l M’) ◦lM’’"
by (simp add: 〈link_wf_3M M M’ M’’ 〉)

hence asm: "(M ◦l M’) ◦l M’’ = (M◦l M’’) ◦l M’"

by simp

have "((M ◦l M’) ◦l M’’), σ →e σ’"

apply (rule link_exec_aux)

apply (rule 〈(M ◦l M’), σ →e σ’ 〉)

using internal_exec_first_step

by blast

from this and asm

have b: "((M ◦l M’’) ◦l M’), σ →e σ’"

by simp

have c: "this_class_lookup σ = Some c ∧ (c ∈ dom M’) "

using internal_exec_first_step

by simp

have d: "this_class_lookup σ’ = Some c’ ∧
(c’ ∈ dom (M ◦l M’’))"

using internal_exec_first_step

by simp

from a b c d internal_exec_first_step

show ?case

using internal_exec.internal_exec_first_step

by simp

next

case (internal_exec_more_steps σ tr σ’ σ’’ c)

from 〈(link_wf_3M M M’ M’’) =⇒
(M ◦l M’’);M’,σ →ei 〈tr 〉 σ’ 〉

and 〈(link_wf_3M M M’ M’’) 〉

have a: "(M ◦l M’’);M’,σ →ei 〈tr 〉 σ’"

by simp

have "M ◦l M’’ = M’’◦l M "

using 〈link_wf_3M M M’ M’’ 〉 link_commute

by blast

hence "(M◦l M’’) ◦l M’ = M’’◦l (M ◦l M’)"

using 〈link_wf_3M M M’ M’’ 〉 link_wf_3M_def link_wf_def

by (metis Int_Un_distrib Un_Int_crazy Un_Int_distrib

Un_commute distrib_imp2 inf.right_idem inf_commute

inf_sup_absorb internal_exec_more_steps.prems link_assoc

link_commute link_dom link_wf_3M_dest(3) link_wf_3M_dest(4)

sup.left_commute sup_assoc sup_bot.left_neutral

sup_bot.right_neutral sup_idem sup_inf_distrib1)

hence "(M◦l M’’) ◦l M’ = (M ◦l M’) ◦lM’’"
by (simp add: 〈link_wf_3M M M’ M’’ 〉)

hence asm: "(M ◦l M’) ◦l M’’ = (M◦l M’’) ◦l M’"

by simp

have "((M ◦l M’) ◦l M’’), σ’ →e σ’’"

apply (rule link_exec_aux)

apply (rule 〈(M ◦l M’), σ’ →e σ’’ 〉)

using 〈link_wf_3M M M’ M’’ 〉

by blast

from this and asm

have b: "((M ◦l M’’) ◦l M’), σ’ →e σ’’"

by simp

have c: "this_class_lookup σ’’ = Some c ∧
(c ∈ dom (M ◦l M’’))"

using internal_exec_more_steps

by simp

from a b c internal_exec.internal_exec_more_steps

show ?case

by simp

qed

lemma visible_exec_linking_1_aux:

" [[(M;M’,σ →e σ’); (link_wf_3M M M’ M’’) ]] =⇒
M; (M’ ◦l M’’), σ →e σ’"

proof (induction rule: visible_exec.induct)

case (visible_exec_intro M M’ σ tr σ’ σ’’ c)

have a: "M; (M’ ◦l M’’), σ →ei 〈tr 〉 σ’"

using 〈 M; M’,σ →ei 〈tr 〉 σ’ 〉

by (simp add: visible_exec_intro internal_linking_1_aux)

have asm: "M ◦l (M’ ◦l M’’) = (M ◦l M’) ◦l M’’"

by (metis link_assoc link_wf_3M_dest(1)

visible_exec_intro.prems)

have "((M ◦l M’) ◦l M’’), σ’ →e σ’’ "

apply (rule link_exec_aux)

apply (rule 〈(M ◦l M’), σ’ →e σ’’ 〉)

using 〈link_wf_3M M M’ M’’ 〉 link_wf_3M_dest

by blast

from this and asm

have b: "(M ◦l (M’ ◦l M’’)), σ’ →e σ’’" by simp

have c: "this_class_lookup σ’’ = Some c ∧
( c ∈ dom (M’ ◦l M’’))"

using 〈this_class_lookup σ’’ = Some c 〉 and 〈c ∈ dom M’ 〉

by simp

from a b c show ?case

using visible_exec.visible_exec_intro

by simp

qed

lemma visible_exec_linking_2_aux:

" [[(M;M’,σ →e σ’); (link_wf_3M M M’ M’’) ]] =⇒
(M ◦l M’’); M’, σ →e σ’"

proof (induction rule: visible_exec.induct)

case (visible_exec_intro M M’ σ tr σ’ σ’’ c)

have a: "(M ◦l M’’);M’,σ →ei 〈tr 〉 σ’"

using 〈M;M’,σ →ei 〈tr 〉 σ’ 〉

by (simp add: 〈link_wf_3M M M’ M’’ 〉 internal_linking_2_aux)

have "M ◦l M’’ = M’’ ◦l M"

using link_commute visible_exec_intro

by auto

hence "(M◦l M’’) ◦l M’ = M’’◦l (M ◦l M’)"

using 〈link_wf_3M M M’ M’’ 〉 link_wf_3M_def link_wf_def

by auto

hence "(M◦l M’’) ◦l M’ = (M ◦l M’) ◦lM’’"
by (simp add: 〈link_wf_3M M M’ M’’ 〉)

hence asm: "(M ◦l M’) ◦l M’’ = (M◦l M’’) ◦l M’"

by simp

have "((M ◦l M’) ◦l M’’), σ’ →e σ’’"

apply (rule link_exec_aux)

apply (rule 〈(M ◦l M’), σ’ →e σ’’ 〉)

using 〈link_wf_3M M M’ M’’ 〉

by blast

from this and asm

have b: "((M ◦l M’’) ◦l M’), σ’ →e σ’’"

by simp

have c: "this_class_lookup σ’’ = Some c ∧ c ∈ dom M’"

using 〈this_class_lookup σ’’ = Some c 〉 and 〈c ∈ dom M’ 〉

by simp

from a b c visible_exec.visible_exec_intro

show ?case

by simp

qed

A.4 Technical Lemmas supporting Adaptation

definition

ident_subst :: "Identifier ⇒ Identifier ⇒ Identifier ⇒ Identifier"

where

"ident_subst x y v = (if v = x then y else v)"

fun

stmt_subst :: "Stmt ⇒ Identifier ⇒ Identifier ⇒ Stmt"

where

"stmt_subst (AssignToField f v) x y =

(AssignToField f (ident_subst x y v))" |

"stmt_subst (ReadFromField v f) x y =

(ReadFromField (ident_subst x y v) f)" |

"stmt_subst (MethodCall v v’ m vs) x y =

(MethodCall (ident_subst x y v) (ident_subst x y v’) m (map (ident_subst x

y) vs))" |

"stmt_subst (NewObject v c vs) x y =

(NewObject (ident_subst x y v) c (map (ident_subst x y) vs))" |

"stmt_subst (Return v) x y =

(Return (ident_subst x y v))"

fun

stmt_idents :: "Stmt ⇒ Identifier set"

where

"stmt_idents (AssignToField f v) = {v}" |

"stmt_idents (ReadFromField v f) = {v}" |

"stmt_idents (MethodCall v v’ m vs) = {v,v’} ∪ (set vs)" |

"stmt_idents (NewObject v c vs) = {v} ∪ (set vs)" |

"stmt_idents (Return v) = {v}"

lemma stmt_subst_idents:

"stmt_idents (stmt_subst s x y) =

((stmt_idents s - {x}) ∪ (if x ∈ stmt_idents s then {y} else {}))"

by (induction rule: stmt_idents.induct,

auto simp: ident_subst_def split: if_splits)

fun

stmts_subst :: "Stmts ⇒ Identifier ⇒ Identifier ⇒ Stmts"

where

"stmts_subst (SingleStmt s) x y =

(SingleStmt (stmt_subst s x y))" |

"stmts_subst (Seq s1 s2) x y =

(Seq (stmt_subst s1 x y) (stmts_subst s2 x y))"

fun

stmts_idents :: "Stmts ⇒ Identifier set"

where

"stmts_idents (SingleStmt s) = (stmt_idents s)" |

"stmts_idents (Seq s1 s2) = (stmt_idents s1 ∪ (stmts_idents s2))"

lemma stmts_subst_idents:

"stmts_idents (stmts_subst s x y) =

((stmts_idents s - {x}) ∪ (if x ∈ stmts_idents s then {y} else {}))"

by (induction rule: stmts_subst.induct,

auto simp: stmt_subst_idents)

fun

stmts_subst_list :: "Stmts ⇒ Identifier list ⇒ Identifier list ⇒ Stmts"

where

"stmts_subst_list s (v#vs) (v’#vs’) =

(stmts_subst_list (stmts_subst s v v’) vs vs’)" |

"stmts_subst_list s (v#vs) [] = undefined" |

"stmts_subst_list s [] (v’#vs’) = undefined" |

"stmts_subst_list s [] [] = s"

definition

stmts_subst_list_wf :: "Stmts ⇒ Identifier list ⇒ Identifier list ⇒ bool"

where

"stmts_subst_list_wf s vs vs’ ≡ (length vs = length vs’) "

lemma stmts_list_idents:

"stmts_subst_list_wf s vs vs’ =⇒
(stmts_idents (stmts_subst_list s vs vs’) ⊆
((stmts_idents s - (set vs)) ∪ (set vs’)))"

proof (induction rule: stmts_subst_list.induct)

case (1 s v vs v’ vs’)

then

have assm1: "stmts_subst_list_wf (stmts_subst s v v’) vs vs’"

and assm2: "stmts_subst_list_wf (stmts_subst s v v’) vs vs’ =⇒
stmts_idents (stmts_subst_list (stmts_subst s v v’) vs vs’)

⊆ stmts_idents (stmts_subst s v v’) - set vs ∪ set vs’"

apply (simp add: stmts_subst_list_wf_def)

by (simp add: "1.IH")

from assm1 and assm2

have assm3: "stmts_idents (stmts_subst_list (stmts_subst s v v’) vs vs’)

⊆ stmts_idents (stmts_subst s v v’) - set vs ∪ set vs’"

by blast

hence assm4: "stmts_idents (stmts_subst_list s (v # vs) (v’ # vs’)) =

stmts_idents (stmts_subst_list (stmts_subst s v v’) vs vs’)"

by auto

from assm1 assm2 assm3 assm4

have assm5: "stmts_idents (stmts_subst s v v’) - set vs ∪ set vs’

⊆ (stmts_idents s - set (v # vs) ∪ set (v’ # vs’))"

using stmts_subst_idents

by auto

from assm1 assm2 assm3 assm4 assm5

have assm6: "stmts_idents (stmts_subst_list s (v # vs) (v’ # vs’))

⊆ (stmts_idents s - set (v # vs) ∪ set (v’ # vs’))"

by blast

thus ?case

by auto

next

case (2 s v vs)

thus ?case

by ( auto simp add: stmts_subst_list_wf_def)

next

case (3 s v’ vs’)

thus ?case

by ( auto simp add: stmts_subst_list_wf_def)

next

case (4 s)

thus ?case

by simp

qed

The fresh_idents X xs is used to generate a list of fresh identifiers where none of the

new identifiers appear in X or xs.

primrec

fresh_idents :: "Identifier set ⇒ Identifier list ⇒ Identifier list"

where

"fresh_idents X [] = []" |

"fresh_idents X (x#xs) = (let v = fresh_nat (X ∪ (set (x#xs))) in

(v # (fresh_idents (X ∪ {v}) xs)))"

lemma fresh_idents_length [simp]:

"length (fresh_idents X xs) = length xs"

apply(induction xs arbitrary: X)

apply clarsimp+

by(metis length_Cons)

lemma fresh_ident_greater:

"finite X =⇒ X ̸= {} =⇒ fresh_nat X > Max X"

unfolding Max_def If_def

by (metis Max_def Max_less_iff fresh_nat_def

last_sorted_list_of_list_is_greatest le_imp_less_Suc)

lemma fresh_idents_greater:

"finite X =⇒ xs ̸= [] =⇒ (∀x ∈ set (fresh_idents X xs). x > Max (X ∪ set

xs))"

apply(induction xs arbitrary: X)

apply simp

apply (clarsimp simp: Let_def)

apply (rule conjI)

using fresh_ident_greater apply simp

apply(rule conjI)

using fresh_ident_greater apply auto[1]

apply clarsimp

apply(subgoal_tac "(insert a (X ∪ set xs)) ̸= {} ∧ finite (insert a (X ∪ set

xs))")

using fresh_ident_greater

proof -

fix a :: nat and xsa :: "nat list"

and Xa :: "nat set" and x :: nat

assume a1:

"x ∈ set (fresh_idents (insert (fresh_nat (insert a (Xa ∪ set xsa))) Xa) xsa)"

assume a2:

"
∧
X. [[finite X; xsa ̸= [] ]] =⇒ ∀x∈set (fresh_idents X xsa). ∀a∈X ∪ set xsa.

a < x"

assume a3: "finite Xa"

assume a4:

"insert a (Xa ∪ set xsa) ̸= {} ∧ finite (insert a (Xa ∪ set xsa))"

have f5:

"∀n. n ∉ set (fresh_idents (insert (fresh_nat (insert a (Xa ∪ set xsa))) Xa)

xsa) ∨
(∀na. na ∉ insert (fresh_nat (insert a (Xa ∪ set xsa))) Xa ∪ set xsa

∨ na < n)"

using a3 a2

by (metis finite.insertI fresh_idents.simps(1) insert_not_empty

mk_disjoint_insert set_empty2)

have f6: "∀n N na. (n::nat) ∉ N ∧ n ̸= na ∨ n ∈ insert na N"

by force

hence "a < fresh_nat (insert a (Xa ∪ set xsa))"

using a4 by (metis (no_types) Max_less_iff fresh_ident_greater)

hence "a < x"

using f5 a1 by fastforce

thus "a < x ∧ (∀n∈Xa ∪ set xsa. n < x)"

using f6 f5 a1 by (metis (no_types) Un_insert_left)

next

fix a xs X x

show

" [[
∧
X. [[finite X; xs ̸= [] ]] =⇒ ∀x∈set (fresh_idents X xs).

∀a∈X ∪ set xs. a < x; finite X;

x ∈ set (fresh_idents (insert (fresh_nat (insert a (X ∪ set xs))) X) xs);∧
X. [[finite X; X ̸= {} ]] =⇒ Max X < fresh_nat X ]]

=⇒ insert a (X ∪ set xs) ̸= {} ∧ finite (insert a (X ∪ set xs))"

by blast

qed

lemma fresh_idents_empty:

"finite X =⇒ (set (fresh_idents X xs) ∩ (X ∪ set xs)) = {}"

proof-

assume "finite X "

hence "(∀x∈set (fresh_idents X xs). x > Max (X ∪ set xs))"

by (metis empty_iff empty_set fresh_idents.simps(1) fresh_idents_greater)

then

show "(set (fresh_idents X xs) ∩ (X ∪ set xs)) = {}"

by (meson Int_emptyI List.finite_set Max_ge 〈finite X 〉 finite_UnI not_le)

qed

lemma fresh_idents_distinct [simp]:

"finite X =⇒ distinct (fresh_idents X xs)"

proof(induction xs arbitrary: X)

case Nil

thus ?case

by clarsimp

next

case (Cons a xs)

show ?case

proof (clarsimp simp: Let_def, rule conjI)

show "fresh_nat (insert a (X ∪ set xs))

∉ set (fresh_idents (insert (fresh_nat (insert a (X ∪ set xs))) X) xs)"

using fresh_idents_empty Cons.prems contra_subsetD

by blast

next

show

"distinct (fresh_idents (insert (fresh_nat (insert a (X ∪ set xs))) X) xs)"

using Cons fresh_idents_empty

by simp

qed

qed

fun

ident_subst_list ::

"Identifier ⇒ Identifier list ⇒ Identifier list ⇒ Identifier"

where

"ident_subst_list x (v#vs) (v’#vs’) = (ident_subst v v’ x)" |

"ident_subst_list x (v#vs) [] = undefined" |

"ident_subst_list x [] (v’#vs’) = undefined" |

"ident_subst_list x [] [] = x"

end

A.5 Technical Lemmas supporting Lemmas
theory Lemmas_Aux

imports Adaptation

begin

Properties of classical logic

lemma aComplement_1:

"Assert_wf A M M’ σ =⇒
(aAnd A (aNot A) M M’ σ) = afalse M M’ σ"

unfolding Assert_wf_def aAnd_def aNot_def afalse_def bopt_def

by auto

lemma aComplement_2:

"Assert_wf A M M’ σ =⇒
(aOr A (aNot A) M M’ σ) = atrue M M’ σ"

unfolding Assert_wf_def aNot_def aOr_def atrue_def bopt_def

by auto

lemma aCommutative_1:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ ]] =⇒
(aOr A A’ M M’ σ) = (aOr A’ A M M’ σ)"

unfolding Assert_wf_def aOr_def bopt_def

by auto

lemma aCommutative_2:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ ]] =⇒
(aAnd A A’ M M’ σ) = (aAnd A’ A M M’ σ)"

unfolding Assert_wf_def aAnd_def bopt_def

by auto

lemma aAssociative_1:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ ]] =⇒
(aOr (aOr A A’) A’’ M M’ σ) = (aOr A (aOr A’ A’’) M M’ σ)"

unfolding aOr_def bopt_def option.case_eq_if

by simp

lemma aAssociative_2:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ ]] =⇒
(aAnd (aAnd A A’) A’’ M M’ σ) = (aAnd A (aAnd A’ A’’) M M’ σ)"

unfolding aAnd_def bopt_def option.case_eq_if

by simp

lemma aDistributive_1:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ; Assert_wf A’’ M M’ σ ]] =⇒
(aAnd (aOr A A’) A’’ M M’ σ) = (aOr (aAnd A A’’) (aAnd A’ A’’) M M’ σ)"

unfolding aAnd_def aOr_def bopt_def option.case_eq_if

by auto

lemma aDistributive_2:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ; Assert_wf A’’ M M’ σ ]] =⇒
(aOr (aAnd A A’) A’’ M M’ σ) = (aAnd (aOr A A’’) (aOr A’ A’’) M M’ σ)"

unfolding aAnd_def aOr_def bopt_def option.case_eq_if

by auto

lemma aDeMorgan_1:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ ]] =⇒
(aNot (aAnd A A’) M M’ σ) = (aOr (aNot A) (aNot A’) M M’ σ)"

unfolding aAnd_def aOr_def aNot_def bopt_def option.case_eq_if

by auto

lemma aDeMorgan_2:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ ]] =⇒
(aNot (aOr A A’) M M’ σ) = (aAnd (aNot A) (aNot A’) M M’ σ)"

unfolding aAnd_def aOr_def aNot_def bopt_def option.case_eq_if

by auto

lemma aUniversal_existential_1:

"Assert_wf A M M’ σ =⇒
aNot (aEx (λx. A )) M M’ σ = aAll (λx. aNot A) M M’ σ"

unfolding Assert_wf_def aNot_def aEx_def aAll_def

by(auto split: option.splits)

lemma aUniversal_existential_2:

"Assert_wf A M M’ σ =⇒
aNot (aAll (λx. A)) M M’ σ = aEx (λx. aNot A) M M’ σ"

unfolding Assert_wf_def aNot_def aEx_def aAll_def

using option.case_eq_if option.simps

by auto

lemma aImplication:

" [[Assert_wf A M M’ σ; Assert_wf A’ M M’ σ]] =⇒
(aAnd A (aImp A A’)) M M’ σ = Some True =⇒

A’ M M’ σ = Some True"

unfolding Assert_wf_def bopt_def aImp_def aAnd_def aImp_def

by auto

lemma aNeverHold:

"Assert_wf A M M’ σ =⇒
(aAnd A (aNot A) M M’ σ) = Some False"

unfolding Assert_wf_def aAnd_def aNot_def bopt_def

by auto

Some lemmas, which could be useful for reasoning about holistic specifications, are

formed and proved in this section. We make the proof details of these lemmas in Appendix

A.5.

lemma object_Unchange_Aux:

"M, σ →e σ’ =⇒ σ = (ϕ#ψ,χ) =⇒
σ’ = (ϕ’#ψ’,χ’) =⇒ finite (dom χ) =⇒
∀a1.((a1 ̸= (this ϕ) ∧ a1 ∈ dom χ ∧ (this ϕ) ∈ dom χ)−→ χ a1 = χ’ a1)"

proof (induction rule: exec.induct)

case (exec_method_call ϕ x y m params stmts a χ C

paramValues M meth ϕ’’ ψ)

thus ?case

by simp

next

case (exec_var_assign ϕ x y stmts M ψ χ)

thus ?case

by simp

next

case (exec_field_assign ϕ y x stmts v χ χ’ M ψ)

thus ?case

by auto

next

case (exec_new ϕ x C params stmts

paramValues M c obj’ a χ χ’ ψ)

hence "a ∉ dom χ"

by simp

thus ?case

using exec_new

by fastforce

next

case (exec_return ϕ x stmts ϕ’ x’ stmts’ M ψ χ)

thus ?case

by simp

qed

lemma Changed_FieldAssign_Aux:

"M, σ →e σ’ =⇒ σ = (ϕ#ψ,χ) =⇒
σ’ = (ϕ’#ψ’,χ’) =⇒ finite (dom χ) =⇒
χ’ (this ϕ) ̸= χ (this ϕ) =⇒ (this ϕ) ∈ dom χ =⇒
∃f x. x ∈ dom (vars ϕ) −→
Some χ’ = this_field_update ϕ χ f (the (vars ϕ x))"

proof (induction rule: exec.induct)

case (exec_method_call ϕ x y m params stmts a χ C

paramValues M meth ϕ’’ ψ)

thus ?case

by simp

next

case (exec_var_assign ϕ x y stmts M ψ χ)

thus ?case

by simp

next

case (exec_field_assign ϕ y x stmts v χ χ’ M ψ)

thus ?case

using ident_lookup.elims Pair_inject

list.inject option.sel

by metis

next

case (exec_new ϕ x C params stmts paramValues

M c obj’ a χ χ’ ψ)

hence "a ∉ dom χ"

by simp

thus ?case

using exec_new

by fastforce

next

case (exec_return ϕ x stmts ϕ’ x’ stmts’ M ψ χ)

thus ?case

by simp

qed

lemma adapt_to_config_Aux:

"finite (dom (vars ϕ)) =⇒
ϕ’’ = adapt_frame ϕ ϕ’ =⇒
w ∈ dom (vars ϕ’’) =⇒
w ∈ dom (vars ϕ) =⇒
(vars ϕ’’ w) = (vars ϕ w)"

unfolding adapt_frame_def Let_def

using Frame.select_convs(2) UnCI

disjoint_iff_not_equal

fresh_idents_empty

map_upds_apply_nontin

by metis

lemma adapt_to_config’_Aux:

"finite (dom (vars ϕ)) =⇒
ϕ’’ = adapt_frame ϕ ϕ’ =⇒
w ∉ dom (vars ϕ) =⇒
w ∈ dom (vars ϕ’’) =⇒
∃v. (v ∈ dom (vars ϕ’) −→

(vars ϕ’’ w = vars ϕ’ v) ∧ w ∉ dom (vars ϕ’))"

unfolding adapt_frame_def Let_def

using Frame.select_convs(2)

fresh_nat_is_fresh

list.simps(8)

map_upds_Nil2

sorted_list_of_set.infinite

by metis

A.6 Lemmas aiding for Holistic assertions in Isabelle/HOL

The object_Unchange is the formalized proof of Lemma 6 in Section 4.1.

lemma object_Unchange:

"M, σ →e σ’ =⇒ σ = (ϕ#ψ,χ) =⇒
σ’ = (ϕ’#ψ’,χ’) =⇒ finite (dom χ) =⇒
∀a1.((a1 ̸= (this ϕ) ∧ a1 ∈ dom χ ∧

(this ϕ) ∈ dom χ)−→ χ a1 = χ’ a1)"

by (simp add: object_Unchange_Aux)

The Changed_FieldAssign is the formalized proof of Lemma 5 in Section 4.1.

lemma Changed_FieldAssign:

"M, σ →e σ’ =⇒ σ = (ϕ#ψ,χ) =⇒
σ’ = (ϕ’#ψ’,χ’) =⇒ finite (dom χ) =⇒
χ’ (this ϕ) ̸= χ (this ϕ) =⇒ (this ϕ) ∈ dom χ =⇒
∃f x. x ∈ dom (vars ϕ) −→
Some χ’ = this_field_update ϕ χ f (the (vars ϕ x))"

using Changed_FieldAssign_Aux by blast

Here is a formalized proof of Lemma 7 mentioned in Section 4.1.

lemma adapt_to_config:

"finite (dom (vars ϕ)) =⇒
ϕ’’ = adapt_frame ϕ ϕ’ =⇒
w ∈ dom (vars ϕ’’) =⇒
w ∈ dom (vars ϕ) =⇒
(vars ϕ’’ w) = (vars ϕ w)"

using adapt_to_config_Aux by blast

Here is a formalized proof of Lemma 8 discussed in Section 4.1.

lemma adapt_to_config’:

"finite (dom (vars ϕ)) =⇒
ϕ’’ = adapt_frame ϕ ϕ’ =⇒
w ∉ dom (vars ϕ) =⇒
w ∈ dom (vars ϕ’’) =⇒
∃v. (v ∈ dom (vars ϕ’) −→
(vars ϕ’’ w = vars ϕ’ v) ∧ w ∉ dom (vars ϕ’))"

by (simp add: adapt_to_config’_Aux)

Lemma 9 is also stated as a formalized proof here.

lemma adapt_to_config_config’:

"finite (dom (vars ϕ)) =⇒

ϕ’’ = adapt_frame ϕ ϕ’ =⇒
w ∈ dom (vars ϕ’’) =⇒

(w ∉ dom (vars ϕ) ∧ (∃v. v ∈ dom (vars ϕ’) −→
vars ϕ’’ w = vars ϕ’ v ∧ w ∉ dom (vars ϕ’)) ∨
(w ∈ dom (vars ϕ) ∧ (vars ϕ’’ w = vars ϕ w)))"

using adapt_to_config’ adapt_to_config

by meson

end

A.7 Partial Proofs of Theorem 3

Theorem 3. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let oi , where i = 1,2 be identifers such that ⌊o1⌋σ = ⌊this⌋σ, ⌊o2⌋σ′ = ⌊this⌋σ′ , o2 ∈ dom(σ)

and ok ∈ dom(σ).

Also, let heaps be χ and χ′ such that χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show that σ |= Next (oi Access ok ) =⇒ (σ |= (o1 Access ok ))∨ (σ |= (o2 Access ok )).

Proof. We consider two cases of oi , including oi = o1 and oi = o2.

In each case, we also call ok = o3, and o3 ∈ dom(σ).

To be convenient, we call LHS of the implication is σ |= Next (oi Access ok ),

and RHS is (σ |= (o1 Access ok ))∨ (σ |= (o2 Access ok )).

• Case oi = o1. We need to show

σ |= Next (o1 Access o3) =⇒ (σ |= (o1 Access o3))∨ (σ |= (o2 Access o3)).

It is proved in Lemma 10.

• Case oi = o2. We need to show

σ |= Next (o2 Access o3) =⇒ (σ |= (o1 Access o3))∨ (σ |= (o2 Access o3)).

It is proved in Lemma 11.

As a result, we have

σ |= Next (oi Access ok ) =⇒ (σ |= (o1 Access ok ))∨ (σ |= (o2 Access ok )).

Lemma 10. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let o1 and o2 be identifers such that ⌊o1⌋σ = ⌊this⌋σ, ⌊o2⌋σ′ = ⌊this⌋σ′ , o2 ∈ dom(σ), and

o3 ∈ dom(σ), χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show that σ |= Next (o1 Access o3) =⇒ (σ |= (o1 Access o3))∨ (σ |= (o2 Access o3)).

Proof. To be convenient, we call LHS of the implication is σ |= Next (o1 Access o3), and

the RHS is (σ |= (o1 Access o3))∨ (σ |= (o2 Access o3)).

From LHS of the implication of the formula, σ |= Next (o1 Access o3) holds if

(σ◁σ′) |= (o1 Access o3).

It is equivalent to

(⌊o1⌋σ◁σ′ = ⌊o3⌋σ◁σ′)∨ (⌊o1. f ⌋σ◁σ′ = ⌊o3⌋σ◁σ′)∨
[(⌊o1⌋σ◁σ′ = ⌊this⌋σ◁σ′)∧ (⌊o3⌋σ◁σ′ = ⌊z′⌋σ◁σ′)],

where z′ appears in (σ◁σ′).cont′.

• Case ⌊o1⌋σ◁σ′ = ⌊o3⌋σ◁σ′ .

We have ⌊o1⌋σ◁σ′ and ⌊o3⌋σ◁σ′ are defined in σ◁σ′.
Also, ⌊o1⌋σ◁σ′ = ⌊o1⌋σ and ⌊o3⌋σ◁σ′ = ⌊o3⌋σ, since o1 and o3 are defined in σ (See

Lemma 7).

Therefore, we have ⌊o1⌋σ = ⌊o3⌋σ in σ that implies o1 Accessø3 in σ.

So, we have RHS of the implication of the formula.

• Case ⌊o1. f ⌋σ◁σ′ = ⌊o3⌋σ◁σ′ , with some field f .

The configuration σ◁σ′ and σ′ use the same heap χ′.
We can rewrite ⌊o1. f ⌋σ◁σ′ = v for some v , and χ′(φ(this)) = (C , f ld M ap), where

f ld M ap( f ) = v , since ⌊o1⌋σ◁σ′ is defined, ⌊o1⌋σ◁σ′ = ⌊o1⌋σ, and ⌊o1⌋σ = ⌊this⌋σ.

On the other hand, ⌊o3⌋σ◁σ′ = ⌊o3⌋σ. Hence, v = ⌊o3⌋σ.

We consider two cases which areχ′(φ(this)) =χ(φ(this)) andχ′(φ(this)) ̸=χ(φ(this)).

– Case χ′(φ(this)) =χ(φ(this)). We have ⌊o1. f ⌋σ = ⌊o3⌋σ obviously.

– Case χ′(φ(this)) ̸=χ(φ(this)).

From the Lemma 5, there exists a field f and an identifier x such that x ∈
dom(σ) such that χ′(φ(this)) = field_update(χ(φ(this), f ,φ(x))). There-

fore, f ld M ap( f ) =φ(x) and we also have φ(x) =φ(o3). We have o1 Access o3

in σ. It implies the RHS of the implication of the formula.

• Case (⌊o1⌋σ◁σ′ = ⌊this⌋σ◁σ′)∧(⌊o3⌋σ◁σ′ = ⌊z⌋σ◁σ′), where z′ appears in (σ◁σ′).cont′.
Note that: We have not finished this case.

Lemma 11. Let σ′ be a next configuration such that M ; M ′,σ→e σ
′.

Let o1 and o2 be identifers such that ⌊o1⌋σ = ⌊this⌋σ, ⌊o2⌋σ′ = ⌊this⌋σ′ , o2 ∈ dom(σ), and

o3 ∈ dom(σ), χ(this) ̸∈ dom(M) and χ′(this) ̸∈ dom(M).

Show that σ |= Next (o2 Access o3) =⇒ (σ |= (o1 Access o3))∨ (σ |= (o2 Access o3)).

Proof. To be convenient, we call LHS of the implication is σ |= Next (o2 Access o3), and

RHS is

(σ |= (o1 Access o3))∨ (σ |= (o2 Access o3)).

From LHS of the implication of the formula, σ |= next (o2 Access o3) holds if

(σ◁σ′) |= (o2 Access o3) (1)

We rewrite (1) as an equivalent form as follows.

(⌊o2⌋σ◁σ′ = ⌊o3⌋σ◁σ′)∨ (⌊o2. f ⌋σ◁σ′ = ⌊o3⌋σ◁σ′)∨
[(⌊o2⌋σ◁σ′ = ⌊this⌋σ◁σ′)∧ (⌊o3⌋σ◁σ′ = ⌊z′⌋σ◁σ′)],

where z′ appears in (σ◁σ′).cont′.

• Case ⌊o2⌋σ◁σ′ = ⌊o3⌋σ◁σ′ .

We have ⌊o2⌋σ◁σ′ and ⌊o3⌋σ◁σ′ are defined in σ◁σ′.
Also, we have ⌊o2⌋σ◁σ′ =φ(o2), and ⌊o3⌋σ◁σ′ =φ(o3), since o2 and o3 are defined in

σ (See Lemma 7). Therefore, we have ⌊o2⌋σ = ⌊o3⌋σ in σ and it implies o2 Access o3

in σ. From that, we have RHS of the implication of the formula.

• Case ⌊o2. f ⌋σ◁σ′ = ⌊o3⌋σ◁σ′ , with some field f .

Moreover,σ◁σ′ andσ′ use the same heapχ′. We can rewrite ⌊o2. f ⌋σ◁σ′ = v for some

v , and χ′(φ(o2)) = (C , f ld M ap), where f ld M ap( f ) = v , since ⌊o2⌋σ◁σ′ is defined,

and φ′(o2) =φ(o2).

Also, ⌊o3⌋σ◁σ′ =φ(o3). Hence, v =φ(o3).

Here, we consider two cases. These are φ(o2) ̸=φ(o1) and φ(o2) =φ(o1).

– Case φ(o2) ̸=φ(o1). It means φ(o2) ̸=φ(this) and from the Lemma 6, we also

have χ′(φ(o2)) =χ(φ(o2)).

Hence, χ(φ(o2)) =φ(o3) in σ, since v =φ(o3). From that, we have o2 Access o3.

It implies RHS of the implication of the formula.

– Case φ(o2) = φ(o1). It means φ(o2) = φ(this). We consider two cases which

are χ′(φ(this)) =χ(φ(this)) and χ′(φ(this)) ̸=χ(φ(this)).

* Case χ′(φ(this)) =χ(φ(this)). We have ⌊o2. f ⌋σ = ⌊o3⌋σ obviously.

* Case χ′(φ(this)) ̸=χ(φ(this)).

From the Lemma 5, there exists a field f and an identifier x such that

x ∈ dom(σ) such that χ′(φ(this)) = field_update(χ(φ(this), f ,φ(x))).

Therefore, f ld M ap( f ) = φ(x) and we also have φ(x) = φ(o3). We have

o2 Access o3 in σ. It implies RHS of the implication of the formula.

• Case (⌊o2⌋σ◁σ′ = ⌊this⌋σ◁σ′)∧(⌊o3⌋σ◁σ′ = ⌊z′⌋σ◁σ′), where z′ appears in (σ◁σ′).cont′.
Note that: We have not finished this case.

